Calculation of the charges of Kerr-Newman BH by the solution phase space method

Written by Kamal Hajian
Institute for Research in Fundamental Sciences (IPM), School of Physics
Based on the paper: K. Hajian and M.M. Sheikh-Jabbari, Phys.Rev. D93 (2016) no.4, 044074 , arXiv:1512.05584.

Isfahan University of Technology, December 28th, 2016.

The Kerr-Newman solution

Initializing the xAct

In[1]:=
```
Block[{Print}, << xAct`xTras`]; Block[{Print}, << xAct`TexAct`];
$DefInfoQ = False;
$UndefInfoQ = False;
$CVVerbose = False;
```

Introducing the manifold, the metric and a chart

In[5]:=
```
DefManifold[M, 4, {α, β, γ, δ, ι, κ, λ, μ, ο, ς, ρ, σ, τ, υ, ω, ν}]
DefMetric[-1, g[-α, -β], CD, PrintAs → "g"]
DefChart[B, M, {0, 1, 2, 3}, {t[], r[], θ[], φ[]}, ChartColor → Blue]
```

This is a command for simplifying any result defined on the chart. Hence, by this command, one can find the explicit components of a given tensor calculated on the background metric. The command is "MySimplify[ ]"

In[8]:=
```
MySimplify1[a_] := ChangeCovD[a, CD, PDB];
MySimplify2[b_] := ToBasis[B]@ToBasis[B]@MySimplify1[b];
MySimplify3[c_] := TraceBasisDummy@MySimplify2[c];
MySimplify4[d_] := ComponentArray@MySimplify3[d];
MySimplify5[e_] := Factor@ToValues@ToValues@ToValues@MySimplify4[e];
MySimplify[f_] := Simplify[MySimplify5[f], TimeConstraint → 1000]
```

Defining the Kerr-Newman metric:

In[14]:=
```
DefConstantSymbol[G, PrintAs → "G"]
DefConstantSymbol[m, PrintAs → "m"]
DefConstantSymbol[a, PrintAs → "a"]
DefConstantSymbol[q, PrintAs → "q"]
$Assumptions = And[r[] ∈ Reals, r[] > 0, θ[] ∈ Reals, 0 < θ[] < π,
   G ∈ Reals, G > 0, m ∈ Reals, m >= 0, a ∈ Reals, a >= 0, q ∈ Reals, q >= 0, G^2 m^2 ≥ a^2 + q^2];
```

The Kerr-Newman metric

In[19]:=
$$\rho2 = r[]^2 + a^2 \, \mathrm{Cos}[\theta[]]^2;$$
$$\Delta = r[]^2 - 2\, G\, m\, r[] + a^2 + q^2;$$
$$f = \frac{2\, G\, m\, r[] - q^2}{\rho2};$$

In[22]:=
```
MatrixForm[TheMetric = {{-(1 - f), 0, 0, -f a Sin[θ[]]^2}, {0, ρ2/Δ, 0, 0},
                        {0, 0, ρ2, 0},
                        {-f a Sin[θ[]]^2, 0, 0, (r[]^2 + a^2 + f a^2 Sin[θ[]]^2) Sin[θ[]]^2}}]
```

Out[22]//MatrixForm=

$$\begin{pmatrix} -1 + \frac{-q^2+2 G m r}{a^2 \mathrm{Cos}[\theta]^2+r^2} & 0 & 0 & -\frac{a\,(-q^2+2 G m r)\,\mathrm{Sin}[\theta]^2}{a^2 \mathrm{Cos}[\theta]^2+r^2} \\ 0 & \frac{a^2 \mathrm{Cos}[\theta]^2+r^2}{a^2+q^2-2 G m r+r^2} & 0 & 0 \\ 0 & 0 & a^2 \mathrm{Cos}[\theta]^2 + r^2 & 0 \\ -\frac{a\,(-q^2+2 G m r)\,\mathrm{Sin}[\theta]^2}{a^2 \mathrm{Cos}[\theta]^2+r^2} & 0 & 0 & \mathrm{Sin}[\theta]^2 \left(a^2 + r^2 + \frac{a^2\,(-q^2+2 G m r)\,\mathrm{Sin}[\theta]^2}{a^2 \mathrm{Cos}[\theta]^2+r^2}\right) \end{pmatrix}$$

In[23]:=
```
MetricInBasis[g, -B, TheMetric];
```

The Kerr-Newman gauge field

In[24]:=
```
DefTensor[A[-α], M]
DefTensor[F[-α, -β], M, Antisymmetric[{-α, -β}]]
```

In[26]:=
```
RuleF = MakeRule[{F[-α, -β], CD[-α]@A[-β] - CD[-β]@A[-α]}, MetricOn → All]
```

Out[26]= $\{\mathrm{HoldPattern}[F^{\underline{\alpha\beta}}] :\to \mathrm{Module}[\{\}, \nabla^{\alpha} A^{\beta} - \nabla^{\beta} A^{\alpha}]\}$

In[27]:=
```
AllComponentValues[A[{-α, -B}], {q r[]/ρ2, 0, 0, -q a r[] Sin[θ[]]^2/ρ2}];
ChangeComponents[A[{α, B}], A[{-ρ, -B}]];
```

Computed $A^{\alpha} \to A_{\beta}\ g^{\alpha\beta}$ in 0.038226 Seconds

Calculating curvature tensors etc:

Calculating generic curvature tensors:

In[29]:=
```
MetricCompute[g, B, All]
```

Calculating some other entities which are not calculated by the command above, but will be necessary in our later simplifications:

In[30]:=
```
AllComponentValues[Christoffel[CD, PDB][{ρ, B}, {σ, B}, {-τ, -B}], MySimplify[g[σ, o] Christoffel[CD, PDB][ρ, -o, -τ]]];
AllComponentValues[Christoffel[CD, PDB][{ρ, B}, {σ, B}, {τ, B}], MySimplify[g[τ, α] g[σ, o] Christoffel[CD, PDB][ρ, -o, -α]]];
AllComponentValues[Christoffel[CD, PDB][{ρ, B}, {-σ, -B}, {τ, B}], MySimplify[g[τ, α] Christoffel[CD, PDB][ρ, -σ, -α]]];
AllComponentValues[Christoffel[CD, PDB][{-ρ, -B}, {-σ, -B}, {τ, B}],
  MySimplify[g[-ρ, -β] g[τ, α] Christoffel[CD, PDB][β, -σ, -α]]];
```

In[34]:=
```
ChangeComponents[RicciCD[{α, B}, {β, B}], RicciCD[{-α, -B}, {-β, -B}]];
```

Computed $R[\nabla]_{\alpha}{}^{\beta} \to g^{\beta\gamma}\, R[\nabla]_{\alpha\gamma}$ in 0.361834 Seconds

Computed $R[\nabla]^{\alpha\beta} \to g^{\alpha\gamma}\, R[\nabla]_{\gamma}{}^{\beta}$ in 0.355302 Seconds

e.o.m

Here, we find the equations of motion through $\delta\left(\sqrt{-g}\ \mathcal{L}\right) = \sqrt{-g}\left[(\text{eom}_g)^{\alpha\beta}\ \delta g_{\alpha\beta} + (\text{eom}_A)^{\alpha}\ \delta A_{\alpha}\right] + a$ surface term

In[35]:=
```
ℒ = 1/(16 π G) (RicciScalarCD[] - F[-α, -β] F[α, β]) /. RuleF;
(VarL[g[-α, -β]][ℒ]) // ToCanonical // ContractMetric // Simplify // ContractMetric // ToCanonical // Factor
```

Out[36]=
$$-\frac{1}{32\,G\,\pi}\Big(2\ R[\nabla]^{\alpha\beta} - g^{\alpha\beta}\ R[\nabla] - 4\left(\nabla^{\alpha} A^{\gamma}\right)\left(\nabla^{\beta} A_{\gamma}\right) + 4\left(\nabla^{\beta} A_{\gamma}\right)\left(\nabla^{\gamma} A^{\alpha}\right) - 4\left(\nabla_{\gamma} A^{\beta}\right)\left(\nabla^{\gamma} A^{\alpha}\right) + 4\left(\nabla^{\alpha} A_{\gamma}\right)\left(\nabla^{\gamma} A^{\beta}\right) - 2\ g^{\alpha\beta}\left(\nabla_{\gamma} A_{\delta}\right)\left(\nabla^{\delta} A^{\gamma}\right) + 2\ g^{\alpha\beta}\left(\nabla_{\delta} A_{\gamma}\right)\left(\nabla^{\delta} A^{\gamma}\right)\Big)$$

In[37]:=
```
eom$g = % * g[-α, -μ] g[-β, -ν] // ContractMetric // ToCanonical // Simplify
```

Out[37]=
$$\frac{1}{32 G \pi}\left(-2 R[\nabla]_{\mu\nu} + g_{\mu\nu}\left(R[\nabla] + 2 (\nabla_\alpha A_\beta)(\nabla^\beta A^\alpha) - 2 (\nabla_\beta A_\alpha)(\nabla^\beta A^\alpha)\right) + 4\left((\nabla_\alpha A_\nu)(\nabla^\alpha A_\mu) - (\nabla^\alpha A_\nu)(\nabla_\mu A_\alpha) - (\nabla^\alpha A_\mu)(\nabla_\nu A_\alpha) + (\nabla_\mu A^\alpha)(\nabla_\nu A_\alpha)\right)\right)$$

In[38]:=
```
eom$A = (VarD[A[-α], CD][ℒ]) // ToCanonical // ContractMetric // ToCanonical // Simplify
```

Out[38]=
$$\frac{-(\nabla_\beta \nabla^\alpha A^\beta) + \nabla_\beta \nabla^\beta A^\alpha}{4 G \pi}$$

We can check that the Kerr-Newman geometry satisfies these equations

In[39]:=
```
eom$g // ToBasis[B] // ToBasis[B] // TraceBasisDummy // ComponentArray // MySimplify
```

Out[39]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}

In[40]:=
```
eom$A // MySimplify
```

Out[40]= {0, 0, 0, 0}

Now, we define rules to impose the on-shell condition wherever we want

In[41]:=
```
Rule$eom$g = MakeRule[Evaluate[{RicciCD[-μ, -ν], (16 π G * (eom$g) + RicciCD[-μ, -ν]) // Simplify}], MetricOn → All]
```

Out[41]=
$$\left\{\text{HoldPattern}\left[R[\nabla]^{\underline{\mu}\underline{\nu}}\right] :\to \text{Module}\left[\{\alpha, \beta\}, \frac{1}{2} g^{\nu\mu} R[\nabla] + 2 (\nabla^\mu A^\alpha)(\nabla^\nu A_\alpha) - 2 (\nabla^\nu A_\alpha)(\nabla^\alpha A^\mu) + 2 (\nabla_\alpha A^\nu)(\nabla^\alpha A^\mu) - 2 (\nabla^\mu A_\alpha)(\nabla^\alpha A^\nu) + g^{\nu\mu} (\nabla_\alpha A_\beta)(\nabla^\beta A^\alpha) - g^{\nu\mu} (\nabla_\beta A_\alpha)(\nabla^\beta A^\alpha)\right]\right\}$$

In[42]:=
```
Rule$eom$A =
  MakeRule[Evaluate[{CD[-β][CD[β][A[α]]], (-4 π G (eom$A) + CD[-β][CD[β][A[α]]]) // FullSimplification[]}], MetricOn → All]
```

Out[42]=
$$\left\{\text{HoldPattern}\left[\nabla_{\underline{\beta}} \nabla^{\underline{\beta}} A^{\underline{\alpha}}\right] :\to \text{Module}\left[\{\gamma\}, \nabla_\gamma \nabla^\alpha A^\gamma\right], \text{HoldPattern}\left[\nabla^{\underline{\beta}} \nabla_{\underline{\beta}} A^{\underline{\alpha}}\right] :\to \text{Module}\left[\{\gamma\}, \nabla_\gamma \nabla^\alpha A^\gamma\right]\right\}$$

To cross check, we can check the vanishing of the equations of motion via on-shell rules defined above:

In[43]:=
```
eom$g /. Rule$eom$g // ToCanonical // FullSimplification[]
```

Out[43]= 0

In[44]:=
```
eom$A /. Rule$eom$A // ToCanonical // FullSimplification[]
```

Out[44]= 0

Charge Calculation

Finding the Θ^μ surface term for the Einstein-Maxwell theory:

We can find the surface term Θ^μ by variating the Lagrangian with respect to the dynamical field

$$\delta\left(\sqrt{-g}\,\mathcal{L}\right) = \sqrt{-g}\left[(\mathrm{eom}_g)^{\alpha\beta}\,\delta g_{\alpha\beta} + (\mathrm{eom}_A)^{\alpha}\,\delta A_\alpha\right] + \sqrt{-g}\,\nabla_\mu \Theta^\mu$$

In[45]:=
```
ExpandPerturbation@Perturbation[Sqrt[-Detg[]] ℒ] // ContractMetric // ToCanonical // Factor
```

Out[45]=
$$-\frac{1}{32\,G\,\pi}\sqrt{-\tilde{\tilde{g}}}\,\Big(2\,\Delta g^{1\alpha\beta}\,R[\nabla]_{\alpha\beta} - \Delta g^{1\alpha}{}_{\alpha}\,R[\nabla] - 2\left(\nabla_\beta\nabla_\alpha\Delta g^{1\alpha\beta}\right) + 2\left(\nabla_\beta\nabla^\beta\Delta g^{1\alpha}{}_{\alpha}\right) - 2\,\Delta g^{1\gamma}{}_{\gamma}\left(\nabla_\alpha A_\beta\right)\left(\nabla^\beta A^\alpha\right) - 8\left(\nabla_\alpha\triangle[A_\beta]\right)\left(\nabla^\beta A^\alpha\right) + 2\,\Delta g^{1\gamma}{}_{\gamma}\left(\nabla_\beta A_\alpha\right)\left(\nabla^\beta A^\alpha\right) - 4\,\Delta g^{1}{}_{\alpha\gamma}\left(\nabla_\beta A^\gamma\right)\left(\nabla^\beta A^\alpha\right) + 8\left(\nabla_\beta\triangle[A_\alpha]\right)\left(\nabla^\beta A^\alpha\right) - 4\,\Delta g^{1}{}_{\beta\gamma}\left(\nabla^\beta A^\alpha\right)\left(\nabla^\gamma A_\alpha\right) + 8\,\Delta g^{1}{}_{\alpha\gamma}\left(\nabla^\beta A^\alpha\right)\left(\nabla^\gamma A_\beta\right)\Big)$$

Subtracting the equation of motion, i.e. calculating the $\delta\left(\sqrt{-g}\,\mathcal{L}\right) - \sqrt{-g}\left[(\mathrm{eom}_g)^{\alpha\beta}\,\delta g_{\alpha\beta} + (\mathrm{eom}_A)^{\alpha}\,\delta A_\alpha\right]$

In[46]:=
```
% - Sqrt[-Detg[]] eom$g * Perturbationg[LI[1], μ, ν] - Sqrt[-Detg[]] eom$A * Perturbation[A[-α]] // ContractMetric // ToCanonical //
  Factor // Simplify // FullSimplify
```

Out[46]=
$$\frac{1}{16\,G\,\pi}\sqrt{-\tilde{\tilde{g}}}\,\Big(\nabla_\beta\nabla_\alpha\Delta g^{1\alpha\beta} + 4\,\triangle[A^\alpha]\left(\nabla_\beta\nabla_\alpha A^\beta - \nabla_\beta\nabla^\beta A_\alpha\right) - \nabla_\beta\nabla^\beta\Delta g^{1\alpha}{}_{\alpha} + 4\left(\nabla_\alpha\triangle[A_\beta] - \nabla_\beta\triangle[A_\alpha]\right)\left(\nabla^\beta A^\alpha\right)\Big)$$

The result is a total derivative. To make sure, subtracting the following total derivative term, would result to zero:

In[47]:=
```
(% - (Sqrt[-Detg[]] * CD[-β]@(CD[-α][Perturbationg[LI[1], α, β]] +
  4 * Perturbation[A[α]] * (CD[-α][A[β]] - CD[β][A[-α]]) -
 CD[β][Perturbationg[LI[1], α, -α]] )) /
 (16 * G * Pi)) // FullSimplification[]
```

Out[47]= 0

Dropping the divergence, the rest would be the Θ^{μ}. So, we make a rule to identify this tensor.

In[48]:=
```
DefTensor[Θ[μ], M]
RuleΘ = MakeRule[{Θ[β], ((CD[-α][Perturbationg[LI[1], α, β]] +
  4 * Perturbation[A[α]] * (CD[-α][A[β]] - CD[β][A[-α]]) -
 CD[β][Perturbationg[LI[1], α, -α]] )) /
 (16 * G * Pi)}, MetricOn → All]
```

Out[49]= $\left\{\text{HoldPattern}\left[\Theta^{\underline{\beta}}\right] :\to \text{Module}\left[\{\alpha\}, -\frac{\triangle\left[A^{\alpha}\right]\left(\nabla^{\beta} A_{\alpha}\right)}{4\,G\,\pi} - \frac{\nabla^{\beta}\Delta g^{1\alpha}{}_{\alpha}}{16\,G\,\pi} + \frac{\triangle\left[A^{\alpha}\right]\left(\nabla_{\alpha} A^{\beta}\right)}{4\,G\,\pi} + \frac{\nabla_{\alpha}\Delta g^{1\alpha\beta}}{16\,G\,\pi}\right]\right\}$

Finding the Noether charge density $Q^{\mu\nu}$ for the E-M theory and the generator ϵ:

In order to find the Noether charge $Q^{\mu\nu}$ associated with the generator $\epsilon = \{\xi^{\mu},\lambda\}$ which is combined of the diffeomorphism generator ξ^{μ} and the gauge transformation λ, first we need to find the Noether current $J_{\epsilon}{}^{\mu} = \Theta^{\mu}(\delta_{\epsilon} g, \delta_{\epsilon} A) - \xi^{\mu}\mathcal{L}$. Then, using the on-shell relation $J_{\epsilon}{}^{\mu} = \nabla_{\nu} Q^{\mu\nu}$ we can find the $Q^{\mu\nu}$.

The vector ξ^{μ} acts on the fields as Lie variaiton. Besides, under the gauge transformation we have $A_{\mu} \to A_{\mu} + \partial_{\mu}\lambda$. Here, we provide rules to imply it:

In[50]:=
```
DefTensor[ξ[μ], M]
DefTensor[lambda[], M, PrintAs → "λ"]
```

In[52]:=
```
Ruledelϵ$g =
 MakeRule[Evaluate[{Perturbationg[LI[1], -α, -β], LieD[ξ[μ], CD][g[-α, -β]] // ContractMetric // ToCanonical}], MetricOn → All]
```

Out[52]= $\left\{\text{HoldPattern}\left[\Delta g^{1\underline{\alpha\beta}}\right] :\to \text{Module}\left[\{\}, \nabla^{\alpha}\xi^{\beta} + \nabla^{\beta}\xi^{\alpha}\right]\right\}$

In[53]:=
```
Ruledeleϵ$A = MakeRule[
  Evaluate[{Perturbation[A[-α]], (LieD[ξ[μ], CD][A[-α]] // ContractMetric // ToCanonical) + CD[-α][lambda[]]}], MetricOn → All]
```

Out[53]= $\{\text{HoldPattern}[\triangle[A^{\underline{\alpha}}]] :\to \text{Module}[\{\beta\}, \nabla^{\alpha}\lambda + A^{\beta}(\nabla^{\alpha}\xi_{\beta}) + \xi^{\beta}(\nabla_{\beta}A^{\alpha})]\}$

Finding the $J_{\epsilon}{}^{\mu}$

In[54]:=
```
(Θ[μ] /. RuleΘ /. Ruledeleϵ$g /. Ruledeleϵ$A) - ξ[μ] ℒ // ContractMetric // ToCanonical // Factor
```

Out[54]= $-\frac{1}{16 G\pi}\Big(R[\nabla]\,\xi^{\mu} - \nabla_{\alpha}\nabla^{\alpha}\xi^{\mu} - \nabla_{\alpha}\nabla^{\mu}\xi^{\alpha} - 4(\nabla_{\alpha}\lambda)(\nabla^{\alpha}A^{\mu}) + 2\,\xi^{\mu}(\nabla_{\alpha}A_{\beta})(\nabla^{\beta}A^{\alpha}) - 2\,\xi^{\mu}(\nabla_{\beta}A_{\alpha})(\nabla^{\beta}A^{\alpha}) - 4\,\xi^{\alpha}(\nabla_{\alpha}A_{\beta})(\nabla^{\beta}A^{\mu}) - 4\,A^{\alpha}(\nabla_{\beta}\xi_{\alpha})(\nabla^{\beta}A^{\mu}) + 4(\nabla_{\alpha}\lambda)(\nabla^{\mu}A^{\alpha}) + 4\,\xi^{\alpha}(\nabla_{\alpha}A_{\beta})(\nabla^{\mu}A^{\beta}) + 4\,A^{\alpha}(\nabla_{\beta}\xi_{\alpha})(\nabla^{\mu}A^{\beta}) + 2(\nabla^{\mu}\nabla_{\alpha}\xi^{\alpha})\Big)$

$J_{\epsilon}{}^{\mu}$ is a divergence term on-shell.

In[55]:=
```
Jϵμ = (% // Expand // FullSimplification[]) /. Rule$eom$g /. Rule$eom$A // Factor // ToCanonical // FullSimplify
```

Out[55]= $\frac{1}{32 G\pi}\Big(-\xi^{\mu}\big(R[\nabla] + 2(\nabla_{\alpha}A_{\beta} - \nabla_{\beta}A_{\alpha})(\nabla^{\beta}A^{\alpha})\big) + 2\big(\nabla_{\alpha}\nabla^{\alpha}\xi^{\mu} + 4(\nabla_{\alpha}\lambda)(\nabla^{\alpha}A^{\mu} - \nabla^{\mu}A^{\alpha}) + 2\big(\xi^{\alpha}(\nabla_{\alpha}A_{\beta} + \nabla_{\beta}A_{\alpha}) + 2\,A^{\alpha}(\nabla_{\beta}\xi_{\alpha})\big)(\nabla^{\beta}A^{\mu} - \nabla^{\mu}A^{\beta}) - \nabla^{\mu}\nabla_{\alpha}\xi^{\alpha}\big)\Big)$

Dropping the divergence, we define the Noether charge density $Q_{\epsilon}{}^{\mu\nu}$

In[56]:=
```
DefTensor[Q[μ, ν], M]
RuleQ =
 MakeRule[{Q[μ, ν], (CD[ν][ξ[μ]] - CD[μ][ξ[ν]] - 4 (CD[μ]@A[ν] - CD[ν]@A[μ]) (A[β] * ξ[-β] + lambda[])) / (16 π G)}, MetricOn → All]
```

Out[57]= $\{\text{HoldPattern}[Q^{\underline{\mu}\underline{\nu}}] :\to \text{Module}[\{\alpha\}, -\frac{\lambda(\nabla^{\mu}A^{\nu})}{4 G\pi} - \frac{A^{\alpha}\,\xi_{\alpha}(\nabla^{\mu}A^{\nu})}{4 G\pi} - \frac{\nabla^{\mu}\xi^{\nu}}{16 G\pi} + \frac{\lambda(\nabla^{\nu}A^{\mu})}{4 G\pi} + \frac{A^{\alpha}\,\xi_{\alpha}(\nabla^{\nu}A^{\mu})}{4 G\pi} + \frac{\nabla^{\nu}\xi^{\mu}}{16 G\pi}]\}$

To cross check, let's check the vanishing of the $J_{\epsilon}{}^{\mu} - \nabla_{\nu}Q_{\epsilon}{}^{\mu\nu}$ on-shell:

In[58]:=
```
((Jϵμ - CD[-ν]@Q[μ, ν]) /. RuleQ /. Rule$eom$g /. Rule$eom$A // FullSimplification[] // ContractMetric) /. Rule$eom$g /.
    Rule$eom$A // ToCanonical // FullSimplify
```

Out[58]= 0

Finding the important 2-form $k^{\mu\nu}$ for the E-M theory and the generator ϵ:

Finding the most important tensor for charge calculations using $\sqrt{-g}\ k_\epsilon{}^{\mu\nu} = \delta(\sqrt{-g}\ Q_\epsilon{}^{\mu\nu}) - \sqrt{-g}\ (\xi^\nu\,\Theta^\mu\text{-}\xi^\mu\,\Theta^\nu)$

In order to find the $\delta(\sqrt{-g}\ Q_\epsilon{}^{\mu\nu})$, we need rules preventing the δ to act on the ϵ

In[59]:=
```
RuleΔξ$0 = MakeRule[{Perturbation[ξ[μ]], 0}, MetricOn → All]
RuleΔλ$0 = MakeRule[{Perturbation[lambda[]], 0}]
```

Out[59]= {HoldPattern[Δ[$\xi^{\underline{\mu}}$]] :→ Module[{}, 0]}

Out[60]= {HoldPattern[Δ[λ]] :→ Module[{}, 0]}

Finding the $\delta(\sqrt{-g}\ Q_\epsilon{}^{\mu\nu})$:

In[61]:=
```
(Sqrt[-Detg[]] * Q[α, β]) /. RuleQ;
δQϵ = (ExpandPerturbation@Perturbation[%] // ContractMetric // ToCanonical // Factor) /. RuleΔξ$0 /. RuleΔλ$0
```

Out[62]=
$$-\frac{1}{32 G \pi}\sqrt{-\tilde{\tilde{g}}}\Big(4\lambda\,\Delta g^{1\gamma}{}_{\gamma}\,(\nabla^{\alpha}A^{\beta})+4A^{\gamma}\,\Delta g^{1\delta}{}_{\delta}\,\xi_{\gamma}\,(\nabla^{\alpha}A^{\beta})+8\Delta[A_{\gamma}]\,\xi^{\gamma}\,(\nabla^{\alpha}A^{\beta})-8\lambda\,\Delta g^{1\beta}{}_{\gamma}\,(\nabla^{\alpha}A^{\gamma})-8A^{\gamma}\,\Delta g^{1\beta}{}_{\delta}\,\xi_{\gamma}\,(\nabla^{\alpha}A^{\delta})+8\lambda\,(\nabla^{\alpha}\Delta[A^{\beta}])+$$
$$8A^{\gamma}\,\xi_{\gamma}\,(\nabla^{\alpha}\Delta[A^{\beta}])+2\xi^{\gamma}\,(\nabla^{\alpha}\Delta g^{1\beta}{}_{\gamma})+\Delta g^{1\gamma}{}_{\gamma}\,(\nabla^{\alpha}\xi^{\beta})-4\lambda\,\Delta g^{1\gamma}{}_{\gamma}\,(\nabla^{\beta}A^{\alpha})-4A^{\gamma}\,\Delta g^{1\delta}{}_{\delta}\,\xi_{\gamma}\,(\nabla^{\beta}A^{\alpha})-8\Delta[A_{\gamma}]\,\xi^{\gamma}\,(\nabla^{\beta}A^{\alpha})+$$
$$8\lambda\,\Delta g^{1\alpha}{}_{\gamma}\,(\nabla^{\beta}A^{\gamma})+8A^{\gamma}\,\Delta g^{1\alpha}{}_{\delta}\,\xi_{\gamma}\,(\nabla^{\beta}A^{\delta})-8\lambda\,(\nabla^{\beta}\Delta[A^{\alpha}])-8A^{\gamma}\,\xi_{\gamma}\,(\nabla^{\beta}\Delta[A^{\alpha}])-2\xi^{\gamma}\,(\nabla^{\beta}\Delta g^{1\alpha}{}_{\gamma})-\Delta g^{1\gamma}{}_{\gamma}\,(\nabla^{\beta}\xi^{\alpha})+$$
$$8\lambda\,\Delta g^{1\beta}{}_{\gamma}\,(\nabla^{\gamma}A^{\alpha})-8\lambda\,\Delta g^{1\alpha}{}_{\gamma}\,(\nabla^{\gamma}A^{\beta})+2\Delta g^{1\beta}{}_{\gamma}\,(\nabla^{\gamma}\xi^{\alpha})-2\Delta g^{1\alpha}{}_{\gamma}\,(\nabla^{\gamma}\xi^{\beta})+8A^{\gamma}\,\Delta g^{1\beta}{}_{\delta}\,\xi_{\gamma}\,(\nabla^{\delta}A^{\alpha})-8A^{\gamma}\,\Delta g^{1\alpha}{}_{\delta}\,\xi_{\gamma}\,(\nabla^{\delta}A^{\beta})\Big)$$

Introducing the $\sqrt{-g}\ k^{\mu\nu}$

In[63]:=
```
δQϵ - 2 Antisymmetrize[Sqrt[-Detg[]] * Θ[α] ξ[β], {α, β}] // FullSimplification[];
k = (% /. RuleΘ) // FullSimplification[] // Factor
```

Out[64]=
$$-\frac{1}{32 G \pi}\sqrt{-\tilde{\tilde{g}}}$$
$$\Big(4\lambda\,\Delta g^{1\gamma}{}_{\gamma}\,(\nabla^{\alpha}A^{\beta})+4A^{\gamma}\,\Delta g^{1\delta}{}_{\delta}\,\xi_{\gamma}\,(\nabla^{\alpha}A^{\beta})+8\Delta[A_{\gamma}]\,\xi^{\gamma}\,(\nabla^{\alpha}A^{\beta})-8\Delta[A^{\gamma}]\,\xi^{\beta}\,(\nabla^{\alpha}A_{\gamma})-8\lambda\,\Delta g^{1\beta}{}_{\gamma}\,(\nabla^{\alpha}A^{\gamma})-8A^{\gamma}\,\Delta g^{1\beta}{}_{\delta}\,\xi_{\gamma}\,(\nabla^{\alpha}A^{\delta})+$$
$$8\lambda\,(\nabla^{\alpha}\Delta[A^{\beta}])+8A^{\gamma}\,\xi_{\gamma}\,(\nabla^{\alpha}\Delta[A^{\beta}])+2\xi^{\gamma}\,(\nabla^{\alpha}\Delta g^{1\beta}{}_{\gamma})-2\xi^{\beta}\,(\nabla^{\alpha}\Delta g^{1\gamma}{}_{\gamma})+\Delta g^{1\gamma}{}_{\gamma}\,(\nabla^{\alpha}\xi^{\beta})-4\lambda\,\Delta g^{1\gamma}{}_{\gamma}\,(\nabla^{\beta}A^{\alpha})-$$
$$4A^{\gamma}\,\Delta g^{1\delta}{}_{\delta}\,\xi_{\gamma}\,(\nabla^{\beta}A^{\alpha})-8\Delta[A_{\gamma}]\,\xi^{\gamma}\,(\nabla^{\beta}A^{\alpha})+8\Delta[A^{\gamma}]\,\xi^{\alpha}\,(\nabla^{\beta}A_{\gamma})+8\lambda\,\Delta g^{1\alpha}{}_{\gamma}\,(\nabla^{\beta}A^{\gamma})+8A^{\gamma}\,\Delta g^{1\alpha}{}_{\delta}\,\xi_{\gamma}\,(\nabla^{\beta}A^{\delta})-$$
$$8\lambda\,(\nabla^{\beta}\Delta[A^{\alpha}])-8A^{\gamma}\,\xi_{\gamma}\,(\nabla^{\beta}\Delta[A^{\alpha}])-2\xi^{\gamma}\,(\nabla^{\beta}\Delta g^{1\alpha}{}_{\gamma})+2\xi^{\alpha}\,(\nabla^{\beta}\Delta g^{1\gamma}{}_{\gamma})-\Delta g^{1\gamma}{}_{\gamma}\,(\nabla^{\beta}\xi^{\alpha})+$$
$$8\Delta[A^{\gamma}]\,\xi^{\beta}\,(\nabla_{\gamma}A^{\alpha})-8\Delta[A^{\gamma}]\,\xi^{\alpha}\,(\nabla_{\gamma}A^{\beta})+2\xi^{\beta}\,(\nabla_{\gamma}\Delta g^{1\alpha\gamma})-2\xi^{\alpha}\,(\nabla_{\gamma}\Delta g^{1\beta\gamma})+8\lambda\,\Delta g^{1\beta}{}_{\gamma}\,(\nabla^{\gamma}A^{\alpha})-$$
$$8\lambda\,\Delta g^{1\alpha}{}_{\gamma}\,(\nabla^{\gamma}A^{\beta})+2\Delta g^{1\beta}{}_{\gamma}\,(\nabla^{\gamma}\xi^{\alpha})-2\Delta g^{1\alpha}{}_{\gamma}\,(\nabla^{\gamma}\xi^{\beta})+8A^{\gamma}\,\Delta g^{1\beta}{}_{\delta}\,\xi_{\gamma}\,(\nabla^{\delta}A^{\alpha})-8A^{\gamma}\,\Delta g^{1\alpha}{}_{\delta}\,\xi_{\gamma}\,(\nabla^{\delta}A^{\beta})\Big)$$

The parametric variations $\hat{\delta}$ g and $\hat{\delta}$ A :

Here, we introduce the parametric variations to be put into the $k^{\mu\nu}$. One can variate the dynamical fields $g_{\mu\nu}$ and A_{μ} with respect to all of the parameters m, a , and q. Nonetheless, using the linearity of the charges in the perturbations, and to speed up the calculations, we variate the dynamical fields with respect to each one of the parameters separately, and sum up the results eventually.

In[65]:=
```
DefTensor[hatdelg$m[-α, -β], M, PrintAs -> "δ̂ₘg"]
DefTensor[hatdelA$m[-α], M, PrintAs -> "δ̂ₘA"]
DefTensor[hatdelg$a[-α, -β], M, PrintAs -> "δ̂ₐg"]
DefTensor[hatdelA$a[-α], M, PrintAs -> "δ̂ₐA"]
DefTensor[hatdelg$q[-α, -β], M, PrintAs -> "δ̂_q g"]
DefTensor[hatdelA$q[-α], M, PrintAs -> "δ̂_q A"]
DefConstantSymbol[δm, PrintAs → "δm"]
DefConstantSymbol[δa, PrintAs → "δa"]
DefConstantSymbol[δq, PrintAs → "δq"]
```

Variation with respect to the parameter m, i.e. $\hat{\delta}_m$ g and $\hat{\delta}_m$ A

In[74]:=
```
MySimplify[g[-α, -β]];
(D[%, m] * δm) // Simplify;
MatrixForm[%]
```

Out[76]//MatrixForm=

$$\begin{pmatrix} \frac{2 G \delta m\, r}{a^2 \mathrm{Cos}[\theta]^2+r^2} & 0 & 0 & -\frac{2 a G \delta m\, r \,\mathrm{Sin}[\theta]^2}{a^2 \mathrm{Cos}[\theta]^2+r^2} \\ 0 & \frac{2 G \delta m\, r\,(a^2 \mathrm{Cos}[\theta]^2+r^2)}{(a^2+q^2+r(-2 G m+r))^2} & 0 & 0 \\ 0 & 0 & 0 & 0 \\ -\frac{2 a G \delta m\, r\,\mathrm{Sin}[\theta]^2}{a^2 \mathrm{Cos}[\theta]^2+r^2} & 0 & 0 & \frac{2 a^2 G \delta m\, r\,\mathrm{Sin}[\theta]^4}{a^2 \mathrm{Cos}[\theta]^2+r^2} \end{pmatrix}$$

In[77]:=
```
% // InputForm
```

Out[77]//InputForm=
```
{{(2*G*δm*r[])/(a^2*Cos[θ[]]^2 + r[]^2), 0, 0, (-2*a*G*δm*r[]*Sin[θ[]]^2)/(a^2*Cos[θ[]]^2 + r[]^2)},
 {0, (2*G*δm*r[]*(a^2*Cos[θ[]]^2 + r[]^2))/(a^2 + q^2 + r[]*(-2*G*m + r[]))^2, 0, 0}, {0, 0, 0, 0},
 {(-2*a*G*δm*r[]*Sin[θ[]]^2)/(a^2*Cos[θ[]]^2 + r[]^2), 0, 0, (2*a^2*G*δm*r[]*Sin[θ[]]^4)/(a^2*Cos[θ[]]^2 + r[]^2)}}
```

In[78]:=
```
AllComponentValues[hatdelg$m[{-α, -B}, {-β, -B}],
  {{(2 * G * δm * r[]) / (a^2 * Cos[θ[]]^2 + r[]^2), 0, 0, (-2 * a * G * δm * r[] * Sin[θ[]]^2) / (a^2 * Cos[θ[]]^2 + r[]^2)},
   {0, (2 * G * δm * r[] * (a^2 * Cos[θ[]]^2 + r[]^2)) / (a^2 + q^2 + r[] * (-2 * G * m + r[]))^2, 0, 0},
   {0, 0, 0, 0}, {(-2 * a * G * δm * r[] * Sin[θ[]]^2) / (a^2 * Cos[θ[]]^2 + r[]^2),
    0, 0, (2 * a^2 * G * δm * r[] * Sin[θ[]]^4) / (a^2 * Cos[θ[]]^2 + r[]^2)}}];
ChangeComponents[hatdelg$m[{α, B}, {-β, -B}], hatdelg$m[{-ρ, -B}, {-β, -B}]];
ChangeComponents[hatdelg$m[{-α, -B}, {β, B}], hatdelg$m[{-ρ, -B}, {-β, -B}]];
ChangeComponents[hatdelg$m[{α, B}, {β, B}], hatdelg$m[{-ρ, -B}, {-β, -B}]];
```

Computed $\hat{\delta}_m g^{\alpha}{}_{\beta} \rightarrow g^{\alpha\gamma}\ \hat{\delta}_m g_{\gamma\beta}$ in 0.261198 Seconds

Computed $\hat{\delta}_m g_{\alpha}{}^{\beta} \rightarrow g^{\beta\gamma}\ \hat{\delta}_m g_{\alpha\gamma}$ in 0.268157 Seconds

Computed $\hat{\delta}_m g_{\alpha}{}^{\beta} \rightarrow g^{\beta\gamma}\ \hat{\delta}_m g_{\alpha\gamma}$ in 0.316126 Seconds

Computed $\hat{\delta}_m g^{\alpha\beta} \rightarrow g^{\alpha\gamma}\ \hat{\delta}_m g_{\gamma}{}^{\beta}$ in 0.514864 Seconds

In[82]:=
```
MySimplify[A[-α]];
(D[%, m] * δm) // Simplify;
MatrixForm[%]
```

Out[84]//MatrixForm=

$$\begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix}$$

In[85]:=
```
% // InputForm
```

Out[85]//InputForm=
```
{0, 0, 0, 0}
```

In[86]:=
```
AllComponentValues[hatdelA$m[{-α, -B}], {0, 0, 0, 0}];
ChangeComponents[hatdelA$m[{α, B}], hatdelA$m[{-ρ, -B}]];
```

Computed $\hat{\delta}_m A^{\alpha} \rightarrow g^{\alpha\beta}\ \hat{\delta}_m A_{\beta}$ in 0.070977 Seconds

In[88]:=
```
RuleδgSm = MakeRule[{Perturbationg[LI[1], -α, -β], hatdelg$m[-α, -β]}, MetricOn → All]
RuleδA$m = MakeRule[{Perturbation[A[-α]], hatdelA$m[-α]}, MetricOn → All]
```

Out[88]= $\left\{\text{HoldPattern}\left[\Delta g^{1\underline{\alpha\beta}}\right] \mapsto \text{Module}\left[\{\},\ \hat{\delta}_m\, g^{\alpha\beta}\right]\right\}$

Out[89]= $\left\{\text{HoldPattern}\left[\Delta\left[A^{\underline{\alpha}}\right]\right] \mapsto \text{Module}\left[\{\},\ \hat{\delta}_m\, A^{\alpha}\right]\right\}$

Variation with respect to the parameter a, i.e. $\hat{\delta}_a$ g and $\hat{\delta}_a$ A

In[90]:=
```
MySimplify[g[-α, -β]];
(D[%, a] * δa) // Simplify;
MatrixForm[%]
```

Out[92]//MatrixForm=

$$\begin{pmatrix} \frac{2\,a\,\delta a\,\text{Cos}[\theta]^2\,(q^2-2\,G\,m\,r)}{(a^2\,\text{Cos}[\theta]^2+r^2)^2} & 0 & 0 & \frac{\delta a\,(q^2-2\,G\,m\,r)\,(-a^2\,\text{Cos}[\theta]^2+r^2)\,\text{Sin}[\theta]^2}{(a^2\,\text{Cos}[\theta]^2+r^2)^2} \\ 0 & \frac{2\,a\,\delta a\,(-r^2+\text{Cos}[\theta]^2\,(q^2+r\,(-2\,G\,m+r)))}{(a^2+q^2+r\,(-2\,G\,m+r))^2} & 0 & 0 \\ 0 & 0 & 2\,a\,\delta a\,\text{Cos}[\theta]^2 & 0 \\ \frac{\delta a\,(q^2-2\,G\,m\,r)\,(-a^2\,\text{Cos}[\theta]^2+r^2)\,\text{Sin}[\theta]^2}{(a^2\,\text{Cos}[\theta]^2+r^2)^2} & 0 & 0 & \frac{2\,a\,\delta a\,\text{Sin}[\theta]^2\,\left(a^4\,\text{Cos}[\theta]^4+2\,a^2\,\text{Cos}[\theta]^2\,r^2+r^2\,\left(r^2-(q^2-2\,G\,m\,r)\,\text{Sin}[\theta]^2\right)\right)}{(a^2\,\text{Cos}[\theta]^2+r^2)^2} \end{pmatrix}$$

In[93]:=
```
% // InputForm
```

Out[93]//InputForm=
```
{{(2*a*δa*Cos[θ[]]^2*(q^2 - 2*G*m*r[]))/(a^2*Cos[θ[]]^2 + r[]^2)^2, 0, 0,
  (δa*(q^2 - 2*G*m*r[])*(-(a^2*Cos[θ[]]^2) + r[]^2)*Sin[θ[]]^2)/(a^2*Cos[θ[]]^2 + r[]^2)^2},
 {0, (2*a*δa*(-r[]^2 + Cos[θ[]]^2*(q^2 + r[]*(-2*G*m + r[]))))/(a^2 + q^2 + r[]*(-2*G*m + r[]))^2, 0, 0},
 {0, 0, 2*a*δa*Cos[θ[]]^2, 0}, {(δa*(q^2 - 2*G*m*r[])*(-(a^2*Cos[θ[]]^2) + r[]^2)*Sin[θ[]]^2)/(a^2*Cos[θ[]]^2 + r[]^2)^2, 0,
  (2*a*δa*Sin[θ[]]^2*(a^4*Cos[θ[]]^4 + 2*a^2*Cos[θ[]]^2*r[]^2 + r[]^2*(r[]^2 - (q^2 - 2*G*m*r[])*Sin[θ[]]^2)))/
   (a^2*Cos[θ[]]^2 + r[]^2)^2}}
```

In[94]:=
```
AllComponentValues[hatdelg$a[{-α, -B}, {-β, -B}], {{(2 * a * δa * Cos[θ[]]^2 * (q^2 - 2 * G * m * r[])) / (a^2 * Cos[θ[]]^2 + r[]^2)^2,
    0, 0, (δa * (q^2 - 2 * G * m * r[]) * (-(a^2 * Cos[θ[]]^2) + r[]^2) * Sin[θ[]]^2) / (a^2 * Cos[θ[]]^2 + r[]^2)^2},
   {0, (2 * a * δa * (-r[]^2 + Cos[θ[]]^2 * (q^2 + r[] * (-2 * G * m + r[])))) / (a^2 + q^2 + r[] * (-2 * G * m + r[]))^2, 0, 0},
   {0, 0, 2 * a * δa * Cos[θ[]]^2, 0},
   {(δa * (q^2 - 2 * G * m * r[]) * (-(a^2 * Cos[θ[]]^2) + r[]^2) * Sin[θ[]]^2) / (a^2 * Cos[θ[]]^2 + r[]^2)^2, 0, 0,
    (2 * a * δa * Sin[θ[]]^2 * (a^4 * Cos[θ[]]^4 + 2 * a^2 * Cos[θ[]]^2 * r[]^2 + r[]^2 * (r[]^2 - (q^2 - 2 * G * m * r[]) * Sin[θ[]]^2))) /
     (a^2 * Cos[θ[]]^2 + r[]^2)^2}}];
ChangeComponents[hatdelg$a[{α, B}, {-β, -B}], hatdelg$a[{-ρ, -B}, {-β, -B}]];
ChangeComponents[hatdelg$a[{-α, -B}, {β, B}], hatdelg$a[{-ρ, -B}, {-β, -B}]];
ChangeComponents[hatdelg$a[{α, B}, {β, B}], hatdelg$a[{-ρ, -B}, {-β, -B}]];
```

Computed $\hat{\delta}_a g^{\alpha}{}_{\beta} \rightarrow g^{\alpha\gamma}\ \hat{\delta}_a g_{\gamma\beta}$ in 0.313714 Seconds

Computed $\hat{\delta}_a g_{\alpha}{}^{\beta} \rightarrow g^{\beta\gamma}\ \hat{\delta}_a g_{\alpha\gamma}$ in 0.330458 Seconds

Computed $\hat{\delta}_a g_{\alpha}{}^{\beta} \rightarrow g^{\beta\gamma}\ \hat{\delta}_a g_{\alpha\gamma}$ in 0.325807 Seconds

Computed $\hat{\delta}_a g^{\alpha\beta} \rightarrow g^{\alpha\gamma}\ \hat{\delta}_a g_{\gamma}{}^{\beta}$ in 0.354060 Seconds

In[98]:=
```
MySimplify[A[-α]];
(D[%, a] * δa) // Simplify;
MatrixForm[%]
```

Out[100]//MatrixForm=

$$\begin{pmatrix} -\frac{2\, a\, q\, \delta a\, \mathrm{Cos}[\theta]^2\, r}{(a^2\, \mathrm{Cos}[\theta]^2+r^2)^2} \\ 0 \\ 0 \\ \frac{q\, \delta a\, r\, (a^2\, \mathrm{Cos}[\theta]^2-r^2)\, \mathrm{Sin}[\theta]^2}{(a^2\, \mathrm{Cos}[\theta]^2+r^2)^2} \end{pmatrix}$$

In[101]:=
```
% // InputForm
```

Out[101]//InputForm=
```
{(-2*a*q*δa*Cos[θ[]]^2*r[])/(a^2*Cos[θ[]]^2  + r[]^2)^2, 0, 0, (q*δa*r[]*(a^2*Cos[θ[]]^2  - r[]^2)*Sin[θ[]]^2)/
  (a^2*Cos[θ[]]^2  + r[]^2)^2}
```

In[102]:=
```
AllComponentValues[hatdelA$a[{-α, -B}], {(-2 * a * q * δa * Cos[θ[]]^2 * r[]) / (a^2 * Cos[θ[]]^2 + r[]^2)^2,
   0, 0, (q * δa * r[] * (a^2 * Cos[θ[]]^2 - r[]^2) * Sin[θ[]]^2) / (a^2 * Cos[θ[]]^2 + r[]^2)^2}];
ChangeComponents[hatdelA$a[{α, B}], hatdelA$a[{-ρ, -B}]];
```

Computed $\hat{\delta}_a A^{\alpha} \to g^{\alpha\beta} \hat{\delta}_a A_{\beta}$ in 0.045129 Seconds

In[104]:=
```
RuleδgSa = MakeRule[{Perturbationg[LI[1], -α, -β], hatdelg$a[-α, -β]}, MetricOn → All]
RuleδA$a = MakeRule[{Perturbation[A[-α]], hatdelA$a[-α]}, MetricOn → All]
```

Out[104]= {HoldPattern[$\Delta g^{1\alpha\beta}$] :→ Module[{}, $\hat{\delta}_a g^{\alpha\beta}$]}

Out[105]= {HoldPattern[$\Delta[A^{\alpha}]$] :→ Module[{}, $\hat{\delta}_a A^{\alpha}$]}

Variation with respect to the parameter q, i.e. $\hat{\delta}_q$ g and $\hat{\delta}_q$ A

In[106]:=
```
MySimplify[g[-α, -β]];
(D[%, q] * δq) // Simplify;
MatrixForm[%]
```

Out[108]//MatrixForm=

$$\begin{pmatrix} -\frac{2 q \delta q}{a^2 \mathrm{Cos}[\theta]^2+r^2} & 0 & 0 & \frac{2 a q \delta q \mathrm{Sin}[\theta]^2}{a^2 \mathrm{Cos}[\theta]^2+r^2} \\ 0 & -\frac{2 q \delta q (a^2 \mathrm{Cos}[\theta]^2+r^2)}{(a^2+q^2+r(-2 G m+r))^2} & 0 & 0 \\ 0 & 0 & 0 & 0 \\ \frac{2 a q \delta q \mathrm{Sin}[\theta]^2}{a^2 \mathrm{Cos}[\theta]^2+r^2} & 0 & 0 & -\frac{2 a^2 q \delta q \mathrm{Sin}[\theta]^4}{a^2 \mathrm{Cos}[\theta]^2+r^2} \end{pmatrix}$$

In[109]:=
```
% // InputForm
```

Out[109]//InputForm=
```
{{(-2*q*δq)/(a^2*Cos[θ[]]^2 + r[]^2), 0, 0, (2*a*q*δq*Sin[θ[]]^2)/(a^2*Cos[θ[]]^2 + r[]^2)},
 {0, (-2*q*δq*(a^2*Cos[θ[]]^2 + r[]^2))/(a^2 + q^2 + r[]*(-2*G*m + r[]))^2, 0, 0}, {0, 0, 0, 0},
 {(2*a*q*δq*Sin[θ[]]^2)/(a^2*Cos[θ[]]^2 + r[]^2), 0, 0, (-2*a^2*q*δq*Sin[θ[]]^4)/(a^2*Cos[θ[]]^2 + r[]^2)}}
```

In[110]:=
```
AllComponentValues[hatdelg$q[{-α, -B}, {-β, -B}],
  {{(-2 * q * δq) / (a^2 * Cos[θ[]]^2 + r[]^2), 0, 0, (2 * a * q * δq * Sin[θ[]]^2) / (a^2 * Cos[θ[]]^2 + r[]^2)},
   {0, (-2 * q * δq * (a^2 * Cos[θ[]]^2 + r[]^2)) / (a^2 + q^2 + r[] * (-2 * G * m + r[]))^2, 0, 0}, {0, 0, 0, 0},
   {(2 * a * q * δq * Sin[θ[]]^2) / (a^2 * Cos[θ[]]^2 + r[]^2), 0, 0, (-2 * a^2 * q * δq * Sin[θ[]]^4) / (a^2 * Cos[θ[]]^2 + r[]^2)}}];
ChangeComponents[hatdelg$q[{α, B}, {-β, -B}], hatdelg$q[{-ρ, -B}, {-β, -B}]];
ChangeComponents[hatdelg$q[{-α, -B}, {β, B}], hatdelg$q[{-ρ, -B}, {-β, -B}]];
ChangeComponents[hatdelg$q[{α, B}, {β, B}], hatdelg$q[{-ρ, -B}, {-β, -B}]];
```

Computed $\hat{\delta}_q g^{\alpha}{}_{\beta} \rightarrow g^{\alpha\gamma}\ \hat{\delta}_q g_{\gamma\beta}$ in 0.208384 Seconds

Computed $\hat{\delta}_q g_{\alpha}{}^{\beta} \rightarrow g^{\beta\gamma}\ \hat{\delta}_q g_{\alpha\gamma}$ in 0.251839 Seconds

Computed $\hat{\delta}_q g_{\alpha}{}^{\beta} \rightarrow g^{\beta\gamma}\ \hat{\delta}_q g_{\alpha\gamma}$ in 0.237071 Seconds

Computed $\hat{\delta}_q g^{\alpha\beta} \rightarrow g^{\alpha\gamma}\ \hat{\delta}_q g_{\gamma}{}^{\beta}$ in 0.299020 Seconds

In[114]:=
```
MySimplify[A[-α]];
(D[%, q] * δq) // Simplify;
MatrixForm[%]
```

Out[116]//MatrixForm=

$$\begin{pmatrix} \frac{\delta q\, r}{a^2 \operatorname{Cos}[\theta]^2 + r^2} \\ 0 \\ 0 \\ -\frac{a\, \delta q\, r \operatorname{Sin}[\theta]^2}{a^2 \operatorname{Cos}[\theta]^2 + r^2} \end{pmatrix}$$

In[117]:=
```
% // InputForm
```

Out[117]//InputForm=
```
{(δq*r[])/(a^2*Cos[θ[]]^2  + r[]^2), 0, 0, -((a*δq*r[]*Sin[θ[]]^2)/(a^2*Cos[θ[]]^2  + r[]^2))}
```

In[118]:=
```
AllComponentValues[hatdelA$q[{-α, -B}],
  {(δq * r[]) / (a^2 * Cos[θ[]]^2 + r[]^2), 0, 0, -((a * δq * r[] * Sin[θ[]]^2) / (a^2 * Cos[θ[]]^2 + r[]^2))}];
ChangeComponents[hatdelA$q[{α, B}], hatdelA$q[{-ρ, -B}]];
```

Computed $\hat{\delta}_q A^{\alpha} \rightarrow g^{\alpha\beta}\ \hat{\delta}_q A_{\beta}$ in 0.052939 Seconds

In[120]:=
```
RuleδgSq = MakeRule[{Perturbationg[LI[1], -α, -β], hatdelg$q[-α, -β]}, MetricOn → All]
RuleδA$q = MakeRule[{Perturbation[A[-α]], hatdelA$q[-α]}, MetricOn → All]
```

Out[120]= $\{\text{HoldPattern}[\Delta g^{1\alpha\beta}] \mapsto \text{Module}[\{\}, \hat{\delta}_q g^{\alpha\beta}]\}$

Out[121]= $\{\text{HoldPattern}[\Delta[A^{\alpha}]] \mapsto \text{Module}[\{\}, \hat{\delta}_q A^{\alpha}]\}$

From here, we concentrate on charge calculations for some of the exact symmetry generators $\eta = \{\zeta, \lambda\}$. Exact symmetries are diff+gauge transformations which their combined action keeps the dynamical fields invariant. So, ζ^{μ} would be *a* Killing vector. For the exact symmetries, the charges would be conserved and independent of the choice of the surfaces of integration.

Mass

Defining the exact symmetry generator for the mass $\eta_M = \{\zeta_\infty, -\Phi_\infty\}$ in which $\boldsymbol{\zeta_\infty = \partial_t + \Omega_\infty \partial_\varphi}$:

Asymptotic angular velocity is equal to $\Omega_\infty = \frac{-g_{t\varphi}}{g_{\varphi\varphi}}\Big|_{r\to\infty}$

In[122]:=
```
DefConstantSymbol[Ω∞]
```

In[123]:=
```
MySimplify[g[-α, -β]];
Limit[-%[[1, 4]]/%[[4, 4]], r[] → ∞] // FullSimplify
```

Out[124]= 0

In[125]:=
```
RuleΩ∞ = MakeRule[{Ω∞, 0}, MetricOn → All]
```

Out[125]= {HoldPattern[Ω∞] ⧴ Module[{}, 0]}

```
In[126]:= DefTensor[ξ∞[α], M]
          AllComponentValues[ξ∞[{α, B}], {1, 0, 0, Ω∞}];
          ChangeComponents[ξ∞[{-α, -B}], ξ∞[{ρ, B}]];
```

Computed $\xi\infty_{\alpha} \to g_{\beta\alpha}\ \xi\infty^{\beta}$ in 0.065517 Seconds

Asymptotic electric poten tials $\Phi_{\infty} = \xi_{\infty}{}^{\mu} A_{\mu} \, |_{r\to\infty}$:

```
In[129]:= DefConstantSymbol[Φ∞]

In[130]:= Limit[MySimplify[ξ∞[-α] A[α]], r[] → ∞] /. RuleΩ∞ // FullSimplify

Out[130]= 0

In[131]:= RuleΦ∞ = MakeRule[{Φ∞, 0}, MetricOn → All]

Out[131]= {HoldPattern[Φ∞] :→ Module[{}, 0]}
```

Rules to identify $\eta_M = \{\partial_t + \Omega_\infty \partial_\varphi,\ -\Phi_\infty\}$:

```
In[132]:= RuleηM$ξ = MakeRule[{ξ[μ], ξ∞[μ]}, MetricOn → All]
          RuleηM$λ = MakeRule[{lambda[], -Φ∞}, MetricOn → All]
```

Out[132]= $\{\text{HoldPattern}[\ \xi^{\mu}_{=}\ ] :\to \text{Module}[\{\},\ \xi\infty^{\mu}]\}$

```
Out[133]= {HoldPattern[λ] :→ Module[{}, -Φ∞]}
```

Let's check to see whether the η_M is an exact symmetry:

```
In[134]:= (LieD[ξ[μ], CD][g[-α, -β]] /. RuleηM$ξ // ContractMetric // ToCanonical) // MySimplify

Out[134]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}

In[135]:= ((LieD[ξ[μ], CD][A[-α]] /. RuleηM$ξ // ContractMetric // ToCanonical) + (CD[-α][lambda[]]) /. RuleηM$λ) // MySimplify

Out[135]= {0, 0, 0, 0}
```

Calculating the variation of the mass with respect to the parameter m, i.e. the $\hat{\delta}_m M$

In[136]:=
```
(k /. RuleηM$ξ /. RuleηM$λ /. Ruleδg$m /. RuleδA$m /. RuleΦ∞ /. RuleΩ∞) // ContractMetric // ToCanonical // FullSimplification[] //
  FullSimplify;
% // ToBasis[B] // ToBasis[B] // ComponentArray;
```

One can use the pull-back of the Hodge dual of the result above to any closed codimension-2 surface surrounding the singularity at the origin. Nonetheless, pull-back to the surfaces of constant time and radius makes the calculations simpler. So, we choose the k^{01} component.

In[138]:=
```
%[[1, 2]] // Factor // Simplify;
% // MySimplify
```

Out[139]= $-\left(\left(\delta m\left(a^4 - a^5\,\Omega\infty - 12\,a^2\,r^2 + 5\,a^3\,\Omega\infty\,r^2 - 16\,r^4 + 12\,a\,\Omega\infty\,r^4 + a^3\,\mathrm{Cos}[4\,\theta]\left(-a + a^2\,\Omega\infty - \Omega\infty\,r^2\right) - 4\,a\,\mathrm{Cos}[2\,\theta]\,r^2\left(a + a^2\,\Omega\infty + 3\,\Omega\infty\,r^2\right)\right)\mathrm{Sin}[\theta]\right) \Big/ \left(16\,\pi\left(a^2 + a^2\,\mathrm{Cos}[2\,\theta] + 2\,r^2\right)^2\right)\right)$

One can integrate the result above on θ and φ, calculated on any arbitrary constant radius r>0. Nonetheless, the r→∞ makes the calculations simpler.

In[140]:=
```
Limit[%, r[] → ∞]
```

Out[140]= $\frac{\delta m\,(4 - 3\,a\,\Omega\infty + 3\,a\,\Omega\infty\,\mathrm{Cos}[2\,\theta])\,\mathrm{Sin}[\theta]}{16\,\pi}$

In[141]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[141]= $\frac{\delta m - a\,\delta m\,\Omega\infty}{2\,\pi}$

The $\hat{\delta}_m M$

In[142]:=
```
Integrate[%, {φ[], 0, 2 π}]
```

Out[142]= $\delta m - a\,\delta m\,\Omega\infty$

In[143]:=
```
δM$m = % /. RuleΦ∞ /. RuleΩ∞ // FullSimplify
```

Out[143]= δm

Calculating the variation of the mass with respect to the parameter a, i.e. the $\hat{\delta}_a M$

In[144]:=
```
(k /. RuleηM$ξ /. RuleηM$λ /. Ruleδg$a /. RuleδA$a /. RuleΦ∞ /. RuleΩ∞) // ContractMetric // ToCanonical // FullSimplification[] //
  FullSimplify;
% // ToBasis[B] // ToBasis[B] // ComponentArray;
```

Again, we choose the pull – back to the constant (t, r) surfaces, i.e. the k^{01} component, for simplicity :

In[146]:=
```
%[[1, 2]] // Factor // FullSimplify;
% // MySimplify
```

Out[147]=

$$-\frac{1}{64 G \pi (a^2 + a^2 \mathrm{Cos}[2\theta] + 2 r^2)^4} \delta a \Big(75 a^7 G m - 5 a^8 G m \Omega\infty + 12 a^7 G m \mathrm{Cos}[6\theta] + 4 a^8 G m \Omega\infty \mathrm{Cos}[6\theta] + a^7 G m \mathrm{Cos}[8\theta] + a^8 G m \Omega\infty \mathrm{Cos}[8\theta] - 75 a^7 r - 104 a^5 q^2 r + 29 a^6 q^2 \Omega\infty r - 12 a^7 \mathrm{Cos}[6\theta] r + 4 a^5 q^2 \mathrm{Cos}[6\theta] r - 16 a^6 q^2 \Omega\infty \mathrm{Cos}[6\theta] r - a^7 \mathrm{Cos}[8\theta] r - a^6 q^2 \Omega\infty \mathrm{Cos}[8\theta] r + 456 a^5 G m r^2 - 61 a^6 G m \Omega\infty r^2 + 12 a^5 G m \mathrm{Cos}[6\theta] r^2 + 32 a^6 G m \Omega\infty \mathrm{Cos}[6\theta] r^2 + a^6 G m \Omega\infty \mathrm{Cos}[8\theta] r^2 - 248 a^5 r^3 + 480 a^3 q^2 r^3 - 208 a^4 q^2 \Omega\infty r^3 - 20 a^5 \mathrm{Cos}[6\theta] r^3 - 40 a^4 q^2 \Omega\infty \mathrm{Cos}[6\theta] r^3 + 720 a^3 G m r^4 - 104 a^4 G m \Omega\infty r^4 + 28 a^4 G m \Omega\infty \mathrm{Cos}[6\theta] r^4 - 272 a^3 r^5 + 192 a q^2 r^5 - 48 a^2 q^2 \Omega\infty r^5 + 192 a G m r^6 + 144 a^2 G m \Omega\infty r^6 - 64 a r^7 + 192 G m \Omega\infty r^8 + 4 a^2 \mathrm{Cos}[4\theta] \big(a^6 G m \Omega\infty + 13 a^5 (G m - r) + 26 a^2 \Omega\infty r^3 (2 q^2 + G m r) + 12 \Omega\infty r^5 (5 q^2 + G m r) + a^4 \Omega\infty r (-7 q^2 + 15 G m r) - 4 a r^3 (14 q^2 - 11 G m r + 7 r^2) - 2 a^3 r (3 q^2 - 23 G m r + 17 r^2)\big) - 4 \mathrm{Cos}[2\theta] \big(a^8 G m \Omega\infty - 29 a^7 (G m - r) + 48 G m \Omega\infty r^8 + 48 a^2 \Omega\infty r^5 (q^2 + G m r) + 4 a^6 \Omega\infty r (-q^2 + 2 G m r) + a^4 \Omega\infty r^3 (-10 q^2 + 7 G m r) + 48 a r^5 (-3 q^2 - 3 G m r + r^2) + 32 a^3 r^3 (-2 q^2 - 7 G m r + 3 r^2) + a^5 r (33 q^2 - 157 G m r + 91 r^2)\big)\Big) \mathrm{Sin}[\theta]$$

One can integrate the result above on θ and φ, on any arbitrary radius r>0. Nonetheless, the r→∞ makes the calculations simpler.

In[148]:=
```
Limit[%, r[] → ∞]
```

Out[148]=

$$-\frac{3 m \delta a \Omega\infty \mathrm{Sin}[\theta]^3}{8 \pi}$$

In[149]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[149]= $-\frac{m\,\delta a\,\Omega\infty}{2\,\pi}$

The $\hat{\delta}_a M$

In[150]:=
```
Integrate[%, {φ[], 0, 2 π}]
```

Out[150]= $-m\,\delta a\,\Omega\infty$

In[151]:=
```
δM$a = % /. RuleΦ∞ /. RuleΩ∞ // FullSimplify
```

Out[151]= 0

Calculating the variation of the mass with respect to the parameter q, using the $k^{\mu\nu}$

In[152]:=
```
(k /. RuleηM$ξ /. RuleηM$λ /. Ruleδg$q /. RuleδA$q /. RuleΩ∞ /. RuleΦ∞) // ContractMetric // ToCanonical // FullSimplification[] //
  FullSimplify;
% // ToBasis[B] // ToBasis[B] // ComponentArray;
```

Again, we choose the pull – back to the constant (t, r) surfaces, i.e. the k^{01} component, for simplicity :

In[154]:=
```
%[[1, 2]] // Factor // Simplify;
% // MySimplify /. RuleΦ∞ /. RuleΩ∞ // Simplify
```

Out[155]= $-\left(\left(a^2\,q\,\delta q\,r\,\left(36\,a^2-14\,a^3\,\Omega\infty+a^3\,\Omega\infty\,\mathrm{Cos}[6\,\theta]+16\,r^2-4\,a\,\Omega\infty\,r^2+2\,a\,\mathrm{Cos}[4\,\theta]\,\left(-2\,a+7\,a^2\,\Omega\infty+10\,\Omega\infty\,r^2\right)+\mathrm{Cos}[2\,\theta]\,\left(32\,a^2-a^3\,\Omega\infty+48\,r^2-16\,a\,\Omega\infty\,r^2\right)\right)\mathrm{Sin}[\theta]\right)\Big/\left(16\,G\,\pi\,\left(a^2+a^2\,\mathrm{Cos}[2\,\theta]+2\,r^2\right)^3\right)\right)$

One can integrate the result above on θ and φ, on any arbitrary radius r>0. Nonetheless, the r→∞ makes the calculations simpler.

In[156]:=
```
Limit[%, r[] → ∞]
```

Out[156]= 0

In[157]:=
```
Integrate[%, {θ[], 0, π}]
```
Out[157]= 0

The $\hat{\delta}_q M$

In[158]:=
```
Integrate[%, {φ[], 0, 2 π}]
```
Out[158]= 0

In[159]:=
```
δM$q = % /. RuleΦ∞ /. RuleΩ∞ // FullSimplify
```
Out[159]= 0

Now we can sum up all of the variations:

In[160]:=
```
δM = δM$m + δM$a + δM$q
```
Out[160]= δm

The result shows that δM is integrable. The integrated result is M=m. The constant of integration has been fixed by setting M=0 for the Minkowski spacttime.

Angular momentum

The exact symmetry generator for the angular momentum $\eta_J = \{-\partial_\varphi, 0\}$

In[161]:=
```
DefTensor[ξ[α], M]
AllComponentValues[ξ[{α, B}], {0, 0, 0, -1}];
ChangeComponents[ξ[{-α, -B}], ξ[{ρ, B}]];
RuleηJ$ξ = MakeRule[{ξ[μ], ξ[μ]}, MetricOn → All]
RuleηJ$λ = MakeRule[{lambda[], 0}, MetricOn → All]
```

Computed $\xi_{\alpha} \to g_{\beta\alpha}\ \xi^{\beta}$ in 0.042714 Seconds

Out[164]= $\left\{\text{HoldPattern}\left[\ \underline{\xi}^{\mu}\ \right] :\to \text{Module}\left[\{\},\ \xi^{\mu}\right]\right\}$

Out[165]= {HoldPattern[λ] :→ Module[{}, 0]}

Let's check to see whether the η_J is an exact symmetry:

In[166]:=
```
(LieD[ξ[μ], CD][g[-α, -β]] /. RuleηJ$ξ // ContractMetric // ToCanonical) // MySimplify
```

Out[166]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}

In[167]:=
```
((LieD[ξ[μ], CD][A[-α]] /. RuleηJ$ξ // ContractMetric // ToCanonical) + (CD[-α][lambda[]]) /. RuleηJ$λ) // MySimplify
```

Out[167]= {0, 0, 0, 0}

All the steps in calculating the variations of the angular momentum using the $k^{\mu\nu}$ are the same as the calculations of the mass. So, we will not repeat the descriptions provided above.

The $\hat{\delta}_m J$

In[168]:=
```
(k /. RuleηJ$ξ /. RuleηJ$λ /. Ruleδg$m /. RuleδA$m) // ContractMetric // ToCanonical // FullSimplification[] // FullSimplify;
% // ToBasis[B] // ToBasis[B] // ComponentArray;
%[[1, 2]] // Factor // Simplify;
% // MySimplify
```

Out[171]= $$-\frac{a\ \delta m\ (a^4 - 3\ a^2\ r^2 - 6\ r^4 + a^2\ \text{Cos}[2\ \theta]\ (a^2 - r^2))\ \text{Sin}[\theta]^3}{4\ \pi\ (a^2 + a^2\ \text{Cos}[2\ \theta] + 2\ r^2)^2}$$

In[172]:=
```
Limit[%, r[] → ∞]
```

Out[172]= $\frac{3\, a\, \delta m\, \mathrm{Sin}[\theta]^3}{8\, \pi}$

In[173]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[173]= $\frac{a\, \delta m}{2\, \pi}$

In[174]:=
```
δJ$m = Integrate[%, {φ[], 0, 2 π}]
```

Out[174]= $a\, \delta m$

The $\hat{\delta}_a J$

In[175]:=
```
(k /. RuleηJ$ξ /. RuleηJ$λ /. Ruleδg$a /. RuleδA$a) // ContractMetric // ToCanonical // FullSimplification[] // FullSimplify;
% // ToBasis[B] // ToBasis[B] // ComponentArray;
%[[1, 2]] // Factor // Simplify;
% // MySimplify
```

Out[178]=

$$-\frac{1}{16\, G\, \pi\, (a^2 + a^2 \mathrm{Cos}[2\,\theta] + 2\, r^2)^4}\, \delta a\, \Big(10\, a^8\, G\, m + a^8\, G\, m\, \mathrm{Cos}[6\,\theta] - 46\, a^6\, q^2\, r - a^6\, q^2\, \mathrm{Cos}[6\,\theta]\, r + 94\, a^6\, G\, m\, r^2 + a^6\, G\, m\, \mathrm{Cos}[6\,\theta]\, r^2 + 168\, a^4\, q^2\, r^3 + 132\, a^4\, G\, m\, r^4 + 144\, a^2\, q^2\, r^5 - 48\, a^2\, G\, m\, r^6 - 96\, G\, m\, r^8 + 2\, a^4\, \mathrm{Cos}[4\,\theta]\, \big(3\, a^4\, G\, m + 2\, r^3\, (-10\, q^2 + 7\, G\, m\, r) + a^2\, r\, (-9\, q^2 + 17\, G\, m\, r)\big) + a^2\, \mathrm{Cos}[2\,\theta]\, \big(15\, a^6\, G\, m + 48\, r^5\, (5\, q^2 + G\, m\, r) + 32\, a^2\, r^3\, (4\, q^2 + 5\, G\, m\, r) + a^4\, r\, (-63\, q^2 + 127\, G\, m\, r)\big)\Big)\, \mathrm{Sin}[\theta]^3$$

In[179]:=
```
Limit[%, r[] → ∞]
```

Out[179]= $\frac{3\, m\, \delta a\, \mathrm{Sin}[\theta]^3}{8\, \pi}$

In[180]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[180]= $\frac{m\,\delta a}{2\,\pi}$

In[181]:=
```
δJ$a = Integrate[%, {φ[], 0, 2 π}]
```

Out[181]= $m\,\delta a$

The $\hat{\delta}_q J$

In[182]:=
```
(k /. RuleηJ$ξ /. RuleηJ$λ /. Ruleδg$q /. RuleδA$q) // ContractMetric // ToCanonical // FullSimplification[] // FullSimplify;
% // ToBasis[B] // ToBasis[B] // ComponentArray;
%[[1, 2]] // Factor // Simplify // MySimplify
```

Out[184]= $-\frac{a^3\,q\,\delta q\,r\,(15\,a^2+a^2\,\mathrm{Cos}[4\,\theta]+12\,r^2+4\,\mathrm{Cos}[2\,\theta]\,(4\,a^2+5\,r^2))\,\mathrm{Sin}[\theta]^3}{4\,G\,\pi\,(a^2+a^2\,\mathrm{Cos}[2\,\theta]+2\,r^2)^3}$

In[185]:=
```
Limit[%, r[] → ∞]
```

Out[185]= 0

In[186]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[186]= 0

In[187]:=
```
δJ$q = Integrate[%, {φ[], 0, 2 π}]
```

Out[187]= 0

Now we can sum up all of the variations:

In[188]:=
```
δJ = δJ$m + δJ$a + δJ$q
```

Out[188]= $m\,\delta a + a\,\delta m$

The result shows that δJ=δ(ma) is a total derivative and hence, it is integrable. The integrated result is J=ma in which the constant of integration has been fixed by the choice of J=0 for the Minkowski spacetime.

Electric Charge

The exact symmetry generator for the electric charge $\epsilon_Q = \{0,\ 1\}$

In[189]:=
```
Undef[ξ]
DefTensor[ξ[α], M]
AllComponentValues[ξ[{α, B}], {0, 0, 0, 0}];
ChangeComponents[ξ[{-α, -B}], ξ[{ρ, B}]];
RuleηQ$ξ = MakeRule[{ξ[μ], ξ[μ]}, MetricOn → All]
RuleηQ$λ = MakeRule[{lambda[], 1}, MetricOn → All]
```

Computed $\xi_\alpha \rightarrow g_{\beta\alpha}\ \xi^\beta$ in 0.040872 Seconds

Out[193]= $\{\text{HoldPattern}[\ \xi^{\underline{\mu}}\ ] :\rightarrow \text{Module}[\{\},\ \xi^\mu]\}$

Out[194]= $\{\text{HoldPattern}[\lambda] :\rightarrow \text{Module}[\{\}, 1]\}$

Let’s check to see whether the η_Q is an exact symmetry:

In[195]:=
```
(LieD[ξ[μ], CD][g[-α, -β]] /. RuleηQ$ξ // ContractMetric // ToCanonical) // MySimplify
```

Out[195]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}

In[196]:=
```
((LieD[ξ[μ], CD][A[-α]] /. RuleηQ$ξ // ContractMetric // ToCanonical) + (CD[-α][lambda[]]) /. RuleηQ$λ) // MySimplify
```
Out[196]= {0, 0, 0, 0}

All the steps in calculating the variations of the electric charge using the $k^{\mu\nu}$ are the same as the calculations of the mass. So, we will not repeat the descriptions provided above.

The $\hat{\delta}_m Q$

In[197]:=
```
(k /. RuleηQ$ξ /. RuleηQ$λ /. Ruleδg$m /. RuleδA$m) // ContractMetric // ToCanonical // FullSimplification[] // FullSimplify;
% // ToBasis[B] // ToBasis[B] // ComponentArray;
%[[1, 2]] // Factor // Simplify;
% // MySimplify
```
Out[200]= 0

In[201]:=
```
Limit[%, r[] → ∞]
```
Out[201]= 0

In[202]:=
```
Integrate[%, {θ[], 0, π}]
```
Out[202]= 0

In[203]:=
```
δQ$m = Integrate[%, {φ[], 0, 2 π}]
```
Out[203]= 0

The $\hat{\delta}_a Q$

In[204]:=
```
(k /. RuleηQ$ξ /. RuleηQ$λ /. Ruleδg$a /. RuleδA$a) // ContractMetric // ToCanonical // FullSimplification[] // FullSimplify;
% // ToBasis[B] // ToBasis[B] // ComponentArray;
%[[1, 2]] // Factor // Simplify;
% // MySimplify
```

Out[207]= $-\frac{a\, q\, \delta a\, r^2 \left(9 a^2 - a^2 \mathrm{Cos}[4\,\theta] + 4 r^2 + 4 \mathrm{Cos}[2\,\theta] \left(2 a^2 + 3 r^2\right)\right) \mathrm{Sin}[\theta]}{2\, G\, \pi \left(a^2 + a^2 \mathrm{Cos}[2\,\theta] + 2 r^2\right)^3}$

In[208]:=
```
Limit[%, r[] → ∞]
```

Out[208]= 0

In[209]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[209]= 0

In[210]:=
```
δQ$a = Integrate[%, {φ[], 0, 2 π}]
```

Out[210]= 0

The $\hat{\delta}_q Q$

In[211]:=
```
(k /. RuleηQ$ξ /. RuleηQ$λ /. Ruleδg$q /. RuleδA$q) // ContractMetric // ToCanonical // FullSimplification[] // FullSimplify;
% // ToBasis[B] // ToBasis[B] // ComponentArray;
%[[1, 2]] // Factor // Simplify;
% // MySimplify
```

Out[214]= $-\frac{\delta q \left(a^2 + a^2 \mathrm{Cos}[2\,\theta] - 2 r^2\right) \left(a^2 + r^2\right) \mathrm{Sin}[\theta]}{2\, G\, \pi \left(a^2 + a^2 \mathrm{Cos}[2\,\theta] + 2 r^2\right)^2}$

In[215]:=
```
Limit[%, r[] → ∞]
```

Out[215]= $\frac{\delta q\, \mathrm{Sin}[\theta]}{4\, G\, \pi}$

In[216]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[216]= $\frac{\delta q}{2\,G\,\pi}$

In[217]:=
```
δQ$q = Integrate[%, {φ[], 0, 2 π}]
```

Out[217]= $\frac{\delta q}{G}$

Now we can sum up all of the variations:

In[218]:=
```
δQ = δQ$m + δQ$a + δQ$q
```

Out[218]= $\frac{\delta q}{G}$

The result shows that $\delta Q = \delta\left(\frac{q}{G}\right)$ is a total derivative and hence, it is integrable. The integrated result is $Q = \frac{q}{G}$ in which the constant of integration has been fixed by the choice of Q=0 for the Minkowski spacetime.

Entropy

In order to identify the exact symmetry generator for the entropy, we need to calculate some entities associated to the horizon :

Horizon radius r_H :

In[219]:=
```
Solve[Δ == 0 /. r[] → r, r]
```

Out[219]= $\left\{\left\{r \to G\,m - \sqrt{-a^2 + G^2\,m^2 - q^2}\right\}, \left\{r \to G\,m + \sqrt{-a^2 + G^2\,m^2 - q^2}\right\}\right\}$

In[220]:=
```
rH = G m + Sqrt[-a^2 + G^2 m^2 - q^2];
```

Horizon angular velocity $\Omega_H = \frac{-g_{t\varphi}}{g_{\varphi\varphi}}\Big|_{r\to r_H}$:

In[221]:=
```
DefConstantSymbol[ΩH]
```

In[222]:=
```
MySimplify[g[-α, -β]];
Limit[(-1 * %[[1, 4]])/%[[4, 4]], r[] → rH] // FullSimplify
```

Out[223]=
$$\frac{a}{-q^2 + 2\,G\,m\left(G\,m + \sqrt{-a^2 + G^2\,m^2 - q^2}\right)}$$

In[224]:=
```
RuleΩH = MakeRule[{ΩH, a / (-q^2 + 2 G m (G m + Sqrt[-a^2 + G^2 m^2 - q^2]))}]
```

Out[224]=
$$\left\{\text{HoldPattern}[\Omega H] \mapsto \text{Module}\left[\{\},\ \frac{a}{-q^2 + 2\,G\,m\left(G\,m + \sqrt{-a^2 + G^2\,m^2 - q^2}\right)}\right]\right\}$$

Horizon Killing vector $\zeta_H = \partial_t + \Omega_H\,\partial_\varphi$:

In[225]:=
```
DefTensor[ζH[α], M]
AllComponentValues[ζH[{α, B}], {1, 0, 0, ΩH}];
ChangeComponents[ζH[{-α, -B}], ζH[{ρ, B}]];
```

Computed $\zeta H_{\alpha} \to g_{\beta\alpha}\ \zeta H^{\beta}$ in 0.047059 Seconds

Horizon electric potentials $\Phi_H = \zeta_H{}^{\mu} A_{\mu}\ \big|_{r\to r_H}$:

In[228]:=
```
DefConstantSymbol[ΦH]
```

In[229]:=
```
(ξH[-α] A[α] // MySimplify) /. r[] → rH /. RuleΩH // FullSimplify
```

Out[229]= $\frac{2 a^2 G m q + q^3 \left(G m - \sqrt{-a^2 + G^2 m^2 - q^2}\right)}{4 a^2 G^2 m^2 + q^4}$

In[230]:=
```
RuleΦH = MakeRule[{ΦH, (2 a^2 G m q + q^3 (G m - Sqrt[-a^2 + G^2 m^2 - q^2]))/(4 a^2 G^2 m^2 + q^4)}, MetricOn → All]
```

Out[230]= $\left\{\text{HoldPattern}[\Phi H] \mapsto \text{Module}\left[\{\}, \frac{2 a^2 G m q}{4 a^2 G^2 m^2 + q^4} + \frac{G m q^3}{4 a^2 G^2 m^2 + q^4} - \frac{q^3 \sqrt{-a^2 + G^2 m^2 - q^2}}{4 a^2 G^2 m^2 + q^4}\right]\right\}$

Finding the Hawking temperature T_H:

Finding the surface gravity κH on the horizon

In[231]:=
```
-1/2 (CD[-μ][ξH[-ν]]) * (CD[μ][ξH[ν]]);
% // MySimplify;
Sqrt[%] /. r[] → rH /. θ[] → π/2 /. RuleΩH // Simplify // Expand // FullSimplify
```

Out[233]= $\sqrt{\left(\frac{1}{(4 a^2 G^2 m^2 + q^4)^2} (a^2 - G^2 m^2 + q^2) \left(4 a^2 G^2 m^2 - 8 G^4 m^4 + 8 G^2 m^2 q^2 - q^4 + 8 G^3 m^3 \sqrt{-a^2 + G^2 m^2 - q^2} - 4 G m q^2 \sqrt{-a^2 + G^2 m^2 - q^2}\right)\right)}$

To make the result simpler, we multiply it by $1 = \frac{(rH^2+a^2)}{(rH^2+a^2)}$

In[234]:=
```
κH = (% * Sqrt[(rH^2 + a^2)^2] // ExpandNumerator // FullSimplify) / (rH^2 + a^2)
```

Out[234]= $\frac{\sqrt{-a^2 + G^2 m^2 - q^2}}{a^2 + \left(G m + \sqrt{-a^2 + G^2 m^2 - q^2}\right)^2}$

Hawking temperature of the horizon, TH = $\frac{\kappa H}{2\pi}$

In[235]:=
```
DefConstantSymbol[TH]
```

In[236]:=
```
RuleTH = MakeRule[{TH, κH/(2 π)}];
```

Defining the exact symmetry for the entropy $\eta_H = \frac{1}{T_H}\{\partial_t + \Omega_H \partial_\varphi, -\Phi_H\}$

In[237]:=
```
RuleηH$ξ = MakeRule[{ξ[μ], 1/TH ξH[μ]}, MetricOn → All];
RuleηH$λ = MakeRule[{lambda[], -ΦH/TH}, MetricOn → All];
```

Let's check to see whether the η_H is an exact symmetry:

In[239]:=
```
(LieD[ξ[μ], CD][g[-α, -β]] /. RuleηH$ξ // ContractMetric // ToCanonical) // MySimplify
```

Out[239]=
```
{{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}
```

In[240]:=
```
((LieD[ξ[μ], CD][A[-α]] /. RuleηH$ξ // ContractMetric // ToCanonical) + (CD[-α][lambda[]]) /. RuleηH$λ) // MySimplify
```

Out[240]=
```
{0, 0, 0, 0}
```

All the steps in calculating the variations of the entropy using the $k^{\mu\nu}$ are the same as the calculations of the mass. So, we will not repeat the descriptions provided above.

The $\hat{\delta}_m S$

In[241]:=
```
(k /. RuleηH$ζ /. RuleηH$λ /. Ruleδg$m /. RuleδA$m) // Factor // ContractMetric // ToCanonical // FullSimplification[] //
 FullSimplify
```

Out[241]=
$$\frac{1}{32\,G\,\pi\,TH}\sqrt{-\tilde{\tilde{g}}}\;\Big(2\,\zeta H^{\gamma}\,\big(-(\nabla^{\alpha}\hat{\delta}_m g^{\beta}{}_{\gamma})+\nabla^{\beta}\hat{\delta}_m g^{\alpha}{}_{\gamma}\big)+\hat{\delta}_m g^{\gamma}{}_{\gamma}\,\big(4\,\Phi H\,(\nabla^{\alpha}A^{\beta})-\nabla^{\alpha}\zeta H^{\beta}-4\,\Phi H\,(\nabla^{\beta}A^{\alpha})+\nabla^{\beta}\zeta H^{\alpha}\big)+$$
$$2\,\Big(4\,\Phi H\,\big(\nabla^{\alpha}\hat{\delta}_m A^{\beta}-\nabla^{\beta}\hat{\delta}_m A^{\alpha}+\hat{\delta}_m g^{\alpha\gamma}\,(\nabla^{\beta}A_{\gamma}-\nabla_{\gamma}A^{\beta})\big)+4\,\hat{\delta}_m A^{\gamma}\,\big(\zeta H_{\gamma}\,(-(\nabla^{\alpha}A^{\beta})+\nabla^{\beta}A^{\alpha})+\zeta H^{\beta}\,(\nabla^{\alpha}A_{\gamma}-\nabla_{\gamma}A^{\alpha})+\zeta H^{\alpha}\,(-(\nabla^{\beta}A_{\gamma})+\nabla_{\gamma}A^{\beta})\big)+$$
$$\zeta H^{\beta}\,(\nabla^{\alpha}\hat{\delta}_m g^{\gamma}{}_{\gamma}-\nabla_{\gamma}\hat{\delta}_m g^{\alpha\gamma})+\zeta H^{\alpha}\,\big(-(\nabla^{\beta}\hat{\delta}_m g^{\gamma}{}_{\gamma})+\nabla_{\gamma}\hat{\delta}_m g^{\beta\gamma}\big)-\hat{\delta}_m g^{\beta\gamma}\,\big(4\,\Phi H\,(\nabla^{\alpha}A_{\gamma})-4\,\Phi H\,(\nabla_{\gamma}A^{\alpha})+\nabla_{\gamma}\zeta H^{\alpha}\big)+\hat{\delta}_m g^{\alpha\gamma}\,(\nabla_{\gamma}\zeta H^{\beta})+$$
$$2\,A^{\gamma}\,\zeta H_{\gamma}\,\Big(\hat{\delta}_m g^{\delta}{}_{\delta}\,(-(\nabla^{\alpha}A^{\beta})+\nabla^{\beta}A^{\alpha})+2\,\big(-(\nabla^{\alpha}\hat{\delta}_m A^{\beta})+\nabla^{\beta}\hat{\delta}_m A^{\alpha}+\hat{\delta}_m g^{\beta\delta}\,(\nabla^{\alpha}A_{\delta}-\nabla_{\delta}A^{\alpha})+\hat{\delta}_m g^{\alpha\delta}\,(-(\nabla^{\beta}A_{\delta})+\nabla_{\delta}A^{\beta})\big)\Big)\Big)\Big)$$

In[242]:=
```
% // ToBasis[B] // ToBasis[B] // ComponentArray;
```

In[243]:=
```
%[[1, 2]] // Factor // Simplify;
% // MySimplify
```

Out[244]=
$$-\Big(\big(\delta m\,\big(a^4-a^5\,\Omega H-12\,a^2\,r^2+5\,a^3\,\Omega H\,r^2-16\,r^4+12\,a\,\Omega H\,r^4+a^3\,\mathrm{Cos}[4\,\theta]\,(-a+a^2\,\Omega H-\Omega H\,r^2)-4\,a\,\mathrm{Cos}[2\,\theta]\,r^2\,(a+a^2\,\Omega H+3\,\Omega H\,r^2)\big)\,\mathrm{Sin}[\theta]\big)\Big/$$
$$\big(16\,\pi\,TH\,(a^2+a^2\,\mathrm{Cos}[2\,\theta]+2\,r^2)^2\big)\Big)$$

In[245]:=
```
Limit[%, r[] → ∞]
```

Out[245]=
$$\frac{\delta m\,(4-3\,a\,\Omega H+3\,a\,\Omega H\,\mathrm{Cos}[2\,\theta])\,\mathrm{Sin}[\theta]}{16\,\pi\,TH}$$

In[246]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[246]=
$$\frac{\delta m-a\,\delta m\,\Omega H}{2\,\pi\,TH}$$

In[247]:=
```
Integrate[%, {φ[], 0, 2 π}]
```

Out[247]=
$$\frac{\delta m-a\,\delta m\,\Omega H}{TH}$$

In[248]:=
```
δS$m = % /. RuleΩH /. RuleTH // FullSimplify
```

Out[248]=
$$-\frac{2\pi\left(a^2+q^2-2\,G\,m\left(G\,m+\sqrt{-a^2+G^2m^2-q^2}\right)\right)\delta m}{\sqrt{-a^2+G^2m^2-q^2}}$$

The $\hat{\delta}_a S$

In[249]:=
```
(k /. RuleηH$ξ /. RuleηH$λ /. Ruleδg$a /. RuleδA$a) // Factor // ContractMetric // ToCanonical // FullSimplification[] //
 FullSimplify
```

Out[249]=
$$\frac{1}{32\,G\,\pi\,TH}\sqrt{-\tilde{\tilde{g}}}\Big(2\,\zeta H^{\gamma}\left(-\left(\nabla^{\alpha}\hat{\delta}_a g^{\beta}{}_{\gamma}\right)+\nabla^{\beta}\hat{\delta}_a g^{\alpha}{}_{\gamma}\right)+\hat{\delta}_a g^{\gamma}{}_{\gamma}\left(4\,\Phi H\left(\nabla^{\alpha}A^{\beta}\right)-\nabla^{\alpha}\zeta H^{\beta}-4\,\Phi H\left(\nabla^{\beta}A^{\alpha}\right)+\nabla^{\beta}\zeta H^{\alpha}\right)+$$
$$2\Big(4\,\Phi H\left(\nabla^{\alpha}\hat{\delta}_a A^{\beta}-\nabla^{\beta}\hat{\delta}_a A^{\alpha}+\hat{\delta}_a g^{\alpha\gamma}\left(\nabla^{\beta}A_{\gamma}-\nabla_{\gamma}A^{\beta}\right)\right)+4\,\hat{\delta}_a A^{\gamma}\left(\zeta H_{\gamma}\left(-\left(\nabla^{\alpha}A^{\beta}\right)+\nabla^{\beta}A^{\alpha}\right)+\zeta H^{\beta}\left(\nabla^{\alpha}A_{\gamma}-\nabla_{\gamma}A^{\alpha}\right)+\zeta H^{\alpha}\left(-\left(\nabla^{\beta}A_{\gamma}\right)+\nabla_{\gamma}A^{\beta}\right)\right)+$$
$$\zeta H^{\beta}\left(\nabla^{\alpha}\hat{\delta}_a g^{\gamma}{}_{\gamma}-\nabla_{\gamma}\hat{\delta}_a g^{\alpha\gamma}\right)+\zeta H^{\alpha}\left(-\left(\nabla^{\beta}\hat{\delta}_a g^{\gamma}{}_{\gamma}\right)+\nabla_{\gamma}\hat{\delta}_a g^{\beta\gamma}\right)-\hat{\delta}_a g^{\beta\gamma}\left(4\,\Phi H\left(\nabla^{\alpha}A_{\gamma}\right)-4\,\Phi H\left(\nabla_{\gamma}A^{\alpha}\right)+\nabla_{\gamma}\zeta H^{\alpha}\right)+\hat{\delta}_a g^{\alpha\gamma}\left(\nabla_{\gamma}\zeta H^{\beta}\right)+$$
$$2\,A^{\gamma}\,\zeta H_{\gamma}\left(\hat{\delta}_a g^{\delta}{}_{\delta}\left(-\left(\nabla^{\alpha}A^{\beta}\right)+\nabla^{\beta}A^{\alpha}\right)+2\left(-\left(\nabla^{\alpha}\hat{\delta}_a A^{\beta}\right)+\nabla^{\beta}\hat{\delta}_a A^{\alpha}+\hat{\delta}_a g^{\beta\delta}\left(\nabla^{\alpha}A_{\delta}-\nabla_{\delta}A^{\alpha}\right)+\hat{\delta}_a g^{\alpha\delta}\left(-\left(\nabla^{\beta}A_{\delta}\right)+\nabla_{\delta}A^{\beta}\right)\right)\right)\Big)\Big)$$

In[250]:=
```
% // ToBasis[B] // ToBasis[B] // ComponentArray;
```

In[251]:=
```
%[[1, 2]] // Factor // Simplify;
% // MySimplify
```

Out[252]=

$$\begin{aligned}&\frac{1}{64\,G\,\pi\,TH\,\left(a^2+a^2\,\mathrm{Cos}[2\,\theta]+2\,r^2\right)^4}\\
&\delta a\,\Big(-75\,a^7\,G\,m+5\,a^8\,G\,m\,\Omega H-12\,a^7\,G\,m\,\mathrm{Cos}[6\,\theta]-4\,a^8\,G\,m\,\Omega H\,\mathrm{Cos}[6\,\theta]-a^7\,G\,m\,\mathrm{Cos}[8\,\theta]-a^8\,G\,m\,\Omega H\,\mathrm{Cos}[8\,\theta]+75\,a^7\,r+\\
&\quad 104\,a^5\,q^2\,r-29\,a^6\,q^2\,\Omega H\,r+12\,a^7\,\mathrm{Cos}[6\,\theta]\,r-4\,a^5\,q^2\,\mathrm{Cos}[6\,\theta]\,r+16\,a^6\,q^2\,\Omega H\,\mathrm{Cos}[6\,\theta]\,r+a^7\,\mathrm{Cos}[8\,\theta]\,r+a^6\,q^2\,\Omega H\,\mathrm{Cos}[8\,\theta]\,r-\\
&\quad 456\,a^5\,G\,m\,r^2+416\,a^5\,q\,\Phi H\,r^2+61\,a^6\,G\,m\,\Omega H\,r^2-12\,a^5\,G\,m\,\mathrm{Cos}[6\,\theta]\,r^2-16\,a^5\,q\,\Phi H\,\mathrm{Cos}[6\,\theta]\,r^2-32\,a^6\,G\,m\,\Omega H\,\mathrm{Cos}[6\,\theta]\,r^2-\\
&\quad a^6\,G\,m\,\Omega H\,\mathrm{Cos}[8\,\theta]\,r^2+248\,a^5\,r^3-480\,a^3\,q^2\,r^3+208\,a^4\,q^2\,\Omega H\,r^3+20\,a^5\,\mathrm{Cos}[6\,\theta]\,r^3+40\,a^4\,q^2\,\Omega H\,\mathrm{Cos}[6\,\theta]\,r^3-720\,a^3\,G\,m\,r^4+\\
&\quad 896\,a^3\,q\,\Phi H\,r^4+104\,a^4\,G\,m\,\Omega H\,r^4-28\,a^4\,G\,m\,\Omega H\,\mathrm{Cos}[6\,\theta]\,r^4+272\,a^3\,r^5-192\,a\,q^2\,r^5+48\,a^2\,q^2\,\Omega H\,r^5-192\,a\,G\,m\,r^6+256\,a\,q\,\Phi H\,r^6-\\
&\quad 144\,a^2\,G\,m\,\Omega H\,r^6+64\,a\,r^7-192\,G\,m\,\Omega H\,r^8-4\,a^2\,\mathrm{Cos}[4\,\theta]\,\Big(a^6\,G\,m\,\Omega H+13\,a^5\,(G\,m-r)+26\,a^2\,\Omega H\,r^3\,\left(2\,q^2+G\,m\,r\right)+12\,\Omega H\,r^5\,\left(5\,q^2+G\,m\,r\right)+\\
&\qquad a^4\,\Omega H\,r\,\left(-7\,q^2+15\,G\,m\,r\right)-4\,a\,r^3\,\left(14\,q^2-11\,G\,m\,r+8\,q\,\Phi H\,r+7\,r^2\right)-2\,a^3\,r\,\left(3\,q^2-23\,G\,m\,r+12\,q\,\Phi H\,r+17\,r^2\right)\Big)+\\
&\quad 4\,\mathrm{Cos}[2\,\theta]\,\Big(a^8\,G\,m\,\Omega H-29\,a^7\,(G\,m-r)+48\,G\,m\,\Omega H\,r^8+48\,a^2\,\Omega H\,r^5\,\left(q^2+G\,m\,r\right)+4\,a^6\,\Omega H\,r\,\left(-q^2+2\,G\,m\,r\right)+a^4\,\Omega H\,r^3\,\left(-10\,q^2+7\,G\,m\,r\right)+\\
&\qquad 48\,a\,r^5\,\left(-3\,q^2-3\,G\,m\,r+4\,q\,\Phi H\,r+r^2\right)+32\,a^3\,r^3\,\left(-2\,q^2-7\,G\,m\,r+8\,q\,\Phi H\,r+3\,r^2\right)+a^5\,r\,\left(33\,q^2-157\,G\,m\,r+132\,q\,\Phi H\,r+91\,r^2\right)\Big)\Big)\,\mathrm{Sin}[\theta]\end{aligned}$$

In[253]:=
```
Limit[%, r[] → ∞]
```

Out[253]=

$$-\frac{3\,m\,\delta a\,\Omega H\,\mathrm{Sin}[\theta]^3}{8\,\pi\,TH}$$

In[254]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[254]=

$$-\frac{m\,\delta a\,\Omega H}{2\,\pi\,TH}$$

In[255]:=
```
Integrate[%, {φ[], 0, 2 π}]
```

Out[255]=

$$-\frac{m\,\delta a\,\Omega H}{TH}$$

In[256]:=
```
δS$a = % /. RuleΩH /. RuleTH // FullSimplify
```

Out[256]=

$$-\frac{2\, a\, m\, \pi\, \delta a}{\sqrt{-a^2 + G^2 m^2 - q^2}}$$

The $\hat{\delta}_q S$

In[257]:=
```
(k /. RuleηH$ζ /. RuleηH$λ /. Ruleδg$q /. RuleδA$q) // Factor // ContractMetric // ToCanonical // FullSimplification[] //
 FullSimplify
```

Out[257]=

$$\frac{1}{32\, G\, \pi\, TH}\sqrt{-\tilde{\tilde{g}}}\, \Big(2\, \zeta H^{\gamma}\, \big(-(\nabla^{\alpha} \hat{\delta}_q g^{\beta}{}_{\gamma}) + \nabla^{\beta} \hat{\delta}_q g^{\alpha}{}_{\gamma}\big) + \hat{\delta}_q g^{\gamma}{}_{\gamma}\, \big(4\, \Phi H\, (\nabla^{\alpha} A^{\beta}) - \nabla^{\alpha} \zeta H^{\beta} - 4\, \Phi H\, (\nabla^{\beta} A^{\alpha}) + \nabla^{\beta} \zeta H^{\alpha}\big) + 2\, \Big(4\, \Phi H\, \big(\nabla^{\alpha} \hat{\delta}_q A^{\beta} - \nabla^{\beta} \hat{\delta}_q A^{\alpha} + \hat{\delta}_q g^{\alpha\gamma}\, (\nabla^{\beta} A_{\gamma} - \nabla_{\gamma} A^{\beta})\big) + 4\, \hat{\delta}_q A^{\gamma}\, \big(\zeta H_{\gamma}\, (-(\nabla^{\alpha} A^{\beta}) + \nabla^{\beta} A^{\alpha}) + \zeta H^{\beta}\, (\nabla^{\alpha} A_{\gamma} - \nabla_{\gamma} A^{\alpha}) + \zeta H^{\alpha}\, (-(\nabla^{\beta} A_{\gamma}) + \nabla_{\gamma} A^{\beta})\big) + \zeta H^{\beta}\, \big(\nabla^{\alpha} \hat{\delta}_q g^{\gamma}{}_{\gamma} - \nabla_{\gamma} \hat{\delta}_q g^{\alpha\gamma}\big) + \zeta H^{\alpha}\, \big(-(\nabla^{\beta} \hat{\delta}_q g^{\gamma}{}_{\gamma}) + \nabla_{\gamma} \hat{\delta}_q g^{\beta\gamma}\big) - \hat{\delta}_q g^{\beta\gamma}\, \big(4\, \Phi H\, (\nabla^{\alpha} A_{\gamma}) - 4\, \Phi H\, (\nabla_{\gamma} A^{\alpha}) + \nabla_{\gamma} \zeta H^{\alpha}\big) + \hat{\delta}_q g^{\alpha\gamma}\, (\nabla_{\gamma} \zeta H^{\beta}) + 2\, A^{\gamma}\, \zeta H_{\gamma}\, \big(\hat{\delta}_q g^{\delta}{}_{\delta}\, (-(\nabla^{\alpha} A^{\beta}) + \nabla^{\beta} A^{\alpha}) + 2\, (-(\nabla^{\alpha} \hat{\delta}_q A^{\beta}) + \nabla^{\beta} \hat{\delta}_q A^{\alpha} + \hat{\delta}_q g^{\beta\delta}\, (\nabla^{\alpha} A_{\delta} - \nabla_{\delta} A^{\alpha}) + \hat{\delta}_q g^{\alpha\delta}\, (-(\nabla^{\beta} A_{\delta}) + \nabla_{\delta} A^{\beta}))\big)\Big)\Big)$$

In[258]:=
```
% // ToBasis[B] // ToBasis[B] // ComponentArray;
```

In[259]:=
```
%[[1, 2]] // Factor // Simplify;
% // MySimplify
```

Out[260]=

$$\frac{1}{16\, G\, \pi\, TH\, (a^2 + a^2 \mathrm{Cos}[2\,\theta] + 2\, r^2)^3}\, \delta q\, \Big(12\, a^6\, \Phi H - 36\, a^4\, q\, r + 14\, a^5\, q\, \Omega H\, r - a^5\, q\, \Omega H\, \mathrm{Cos}[6\,\theta]\, r + 12\, a^4\, \Phi H\, r^2 - 16\, a^2\, q\, r^3 + 4\, a^3\, q\, \Omega H\, r^3 - 32\, a^2\, \Phi H\, r^4 - 32\, \Phi H\, r^6 + a^2\, \mathrm{Cos}[2\,\theta]\, \big(16\, a^4\, \Phi H + a^3\, q\, \Omega H\, r - 48\, q\, r^3 + 16\, a\, q\, \Omega H\, r^3 + 16\, a^2\, r\, (-2\, q + \Phi H\, r)\big) + 2\, a^3\, \mathrm{Cos}[4\,\theta]\, \big(2\, a^3\, \Phi H - 7\, a^2\, q\, \Omega H\, r - 10\, q\, \Omega H\, r^3 + 2\, a\, r\, (q + \Phi H\, r)\big)\Big)\, \mathrm{Sin}[\theta]$$

In[261]:=
```
Limit[%, r[] → ∞]
```

Out[261]=

$$-\frac{\delta q\, \Phi H\, \mathrm{Sin}[\theta]}{4\, G\, \pi\, TH}$$

In[262]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[262]= $-\frac{\delta q\ \Phi H}{2\ G\ \pi\ TH}$

In[263]:=
```
Integrate[%, {φ[], 0, 2 π}]
```

Out[263]= $-\frac{\delta q\ \Phi H}{G\ TH}$

In[264]:=
```
δS$q = % /. RuleΩH /. RuleTH /. RuleΦH // FullSimplify
```

Out[264]= $-\frac{2\ \pi\ q\left(1+\frac{G\ m}{\sqrt{-a^2+G^2\ m^2-q^2}}\right)\delta q}{G}$

Now we can sum up all of the variations:

In[265]:=
```
δS = δS$m + δS$a + δS$q
```

Out[265]= $-\frac{2\ a\ m\ \pi\ \delta a}{\sqrt{-a^2+G^2\ m^2-q^2}}-\frac{2\ \pi\left(a^2+q^2-2\ G\ m\left(G\ m+\sqrt{-a^2+G^2\ m^2-q^2}\right)\right)\delta m}{\sqrt{-a^2+G^2\ m^2-q^2}}-\frac{2\ \pi\ q\left(1+\frac{G\ m}{\sqrt{-a^2+G^2\ m^2-q^2}}\right)\delta q}{G}$

The result shows that $\delta S=\delta\left(\frac{4\pi(rH^2+a^2)}{4G}\right)$ is a total derivative and hence, it is integrable. To check this claim:

In[266]:=
```
(δS - ((D[4 π (rH^2 + a^2) / (4 G), m] * δm) + (D[4 π (rH^2 + a^2) / (4 G), a] * δa) + (D[4 π (rH^2 + a^2) / (4 G), q] * δq))) // FullSimplify
```

Out[266]= 0

The integrated result is $S=\frac{4\pi(rH^2+a^2)}{4G}$ in which the constant of integration has been fixed by the choice of S=0 for the Minkowski spacetime.

First law of thermodynamics

In the “solution phase space method,” the first law originates from the local identity between the generators of entropy and other charges: $\eta_H = \frac{1}{T_H}(\eta_M - (\Omega_H - \Omega_\infty)\,\eta_J - (\Phi_H - \Phi_\infty)\,\eta_Q)$. By the linearity of charge variations in their generators, one easily proves the first law as: $\delta S_H = \frac{1}{T_H}(\delta M - (\Omega_H - \Omega_\infty)\,\delta J - (\Phi_H - \Phi_\infty)\,\delta Q)$. To cross check:

In[267]:=
```
(δS - 1/TH (δM - (ΩH - Ω∞) δJ - (ΦH - Φ∞) δQ)) /. RuleΦH /. RuleΦ∞ /. RuleΩH /. RuleΩ∞ /. RuleTH // FullSimplify
```

Out[267]= 0

For a review on the “solution phase space method, the papers below can be refered to:

1) K. Hajian, Gen.Rel.Grav. 48 (2016) no.8, 114, arXiv:1602.05575 [gr-qc].
2) M. Ghodrati, K. Hajian, M.R. Setare, arXiv:1606.04353 , Eur. Phys. J. C. (2016), 76:701.